(** This file collects definitions and theorems on sets. Most
importantly, it implements some tactics to automatically solve goals involving
sets. *)
From stdpp Require Export orders list list_numbers.
From stdpp Require Import finite.
From stdpp Require Import options.

(* FIXME: This file needs a 'Proof Using' hint, but they need to be set
locally (or things moved out of sections) as no default works well enough. *)
Unset Default Proof Using.

(* Higher precedence to make sure these instances are not used for other types
with an [ElemOf] instance, such as lists. *)
Instance set_equiv `{ElemOf A C} : Equiv C | 20 := λ X Y,
  ∀ x, x ∈ X ↔ x ∈ Y.
Instance set_subseteq `{ElemOf A C} : SubsetEq C | 20 := λ X Y,
  ∀ x, x ∈ X → x ∈ Y.
Instance set_disjoint `{ElemOf A C} : Disjoint C | 20 := λ X Y,
  ∀ x, x ∈ X → x ∈ Y → False.
Typeclasses Opaque set_equiv set_subseteq set_disjoint.

(** * Setoids *)
Section setoids_simple.
  Context `{SemiSet A C}.

  Global Instance set_equiv_equivalence : Equivalence (≡@{C}).
  Proof.
    split.
    - done.
    - intros X Y ? x. by symmetry.
    - intros X Y Z ?? x; by trans (x ∈ Y).
  Qed.
  Global Instance singleton_proper : Proper ((=) ==> (≡@{C})) singleton.
  Proof. apply _. Qed.
  Global Instance elem_of_proper : Proper ((=) ==> (≡) ==> iff) (∈@{C}) | 5.
  Proof. by intros x ? <- X Y. Qed.
  Global Instance disjoint_proper: Proper ((≡) ==> (≡) ==> iff) (##@{C}).
  Proof.
    intros X1 X2 HX Y1 Y2 HY; apply forall_proper; intros x. by rewrite HX, HY.
  Qed.
  Global Instance union_proper : Proper ((≡) ==> (≡) ==> (≡@{C})) union.
  Proof. intros X1 X2 HX Y1 Y2 HY x. rewrite !elem_of_union. f_equiv; auto. Qed.
  Global Instance union_list_proper: Proper ((≡) ==> (≡@{C})) union_list.
  Proof. by induction 1; simpl; try apply union_proper. Qed.
  Global Instance subseteq_proper : Proper ((≡@{C}) ==> (≡@{C}) ==> iff) (⊆).
  Proof.
    intros X1 X2 HX Y1 Y2 HY. apply forall_proper; intros x. by rewrite HX, HY.
  Qed.
End setoids_simple.

Section setoids.
  Context `{Set_ A C}.

  (** * Setoids *)
  Global Instance intersection_proper :
    Proper ((≡) ==> (≡) ==> (≡@{C})) intersection.
  Proof.
    intros X1 X2 HX Y1 Y2 HY x. by rewrite !elem_of_intersection, HX, HY.
  Qed.
  Global Instance difference_proper :
     Proper ((≡) ==> (≡) ==> (≡@{C})) difference.
  Proof.
    intros X1 X2 HX Y1 Y2 HY x. by rewrite !elem_of_difference, HX, HY.
  Qed.
End setoids.

Section setoids_monad.
  Context `{MonadSet M}.

  Global Instance set_fmap_proper {A B} :
    Proper (pointwise_relation _ (=) ==> (≡) ==> (≡)) (@fmap M _ A B).
  Proof.
    intros f1 f2 Hf X1 X2 HX x. rewrite !elem_of_fmap. f_equiv; intros z.
    by rewrite HX, Hf.
  Qed.
  Global Instance set_bind_proper {A B} :
    Proper (pointwise_relation _ (≡) ==> (≡) ==> (≡)) (@mbind M _ A B).
  Proof.
    intros f1 f2 Hf X1 X2 HX x. rewrite !elem_of_bind. f_equiv; intros z.
    by rewrite HX, (Hf z).
  Qed.
  Global Instance set_join_proper {A} :
    Proper ((≡) ==> (≡)) (@mjoin M _ A).
  Proof.
    intros X1 X2 HX x. rewrite !elem_of_join. f_equiv; intros z. by rewrite HX.
  Qed.
End setoids_monad.

(** * Tactics *)
(** The tactic [set_unfold] transforms all occurrences of [(∪)], [(∩)], [(∖)],
[(<$>)], [∅], [{[_]}], [(≡)], and [(⊆)] into logically equivalent propositions
involving just [∈]. For example, [A → x ∈ X ∪ ∅] becomes [A → x ∈ X ∨ False].

This transformation is implemented using type classes instead of setoid
rewriting to ensure that we traverse each term at most once and to be able to
deal with occurences of the set operations under binders. *)
Class SetUnfold (P Q : Prop) := { set_unfold : P ↔ Q }.
Arguments set_unfold _ _ {_} : assert.
Global Hint Mode SetUnfold + - : typeclass_instances.

(** The class [SetUnfoldElemOf] is a more specialized version of [SetUnfold]
for propositions of the shape [x ∈ X] to improve performance. *)
Class SetUnfoldElemOf `{ElemOf A C} (x : A) (X : C) (Q : Prop) :=
  { set_unfold_elem_of : x ∈ X ↔ Q }.
Arguments set_unfold_elem_of {_ _ _} _ _ _ {_} : assert.
Global Hint Mode SetUnfoldElemOf + + + - + - : typeclass_instances.

Instance set_unfold_elem_of_default `{ElemOf A C} (x : A) (X : C) :
  SetUnfoldElemOf x X (x ∈ X) | 1000.
Proof. done. Qed.
Instance set_unfold_elem_of_set_unfold `{ElemOf A C} (x : A) (X : C) Q :
  SetUnfoldElemOf x X Q → SetUnfold (x ∈ X) Q.
Proof. by destruct 1; constructor. Qed.

Class SetUnfoldSimpl (P Q : Prop) := { set_unfold_simpl : SetUnfold P Q }.
Global Hint Extern 0 (SetUnfoldSimpl _ _) => csimpl; constructor : typeclass_instances.

Instance set_unfold_default P : SetUnfold P P | 1000. done. Qed.
Definition set_unfold_1 `{SetUnfold P Q} : P → Q := proj1 (set_unfold P Q).
Definition set_unfold_2 `{SetUnfold P Q} : Q → P := proj2 (set_unfold P Q).

Lemma set_unfold_impl P Q P' Q' :
  SetUnfold P P' → SetUnfold Q Q' → SetUnfold (P → Q) (P' → Q').
Proof. constructor. by rewrite (set_unfold P P'), (set_unfold Q Q'). Qed.
Lemma set_unfold_and P Q P' Q' :
  SetUnfold P P' → SetUnfold Q Q' → SetUnfold (P ∧ Q) (P' ∧ Q').
Proof. constructor. by rewrite (set_unfold P P'), (set_unfold Q Q'). Qed.
Lemma set_unfold_or P Q P' Q' :
  SetUnfold P P' → SetUnfold Q Q' → SetUnfold (P ∨ Q) (P' ∨ Q').
Proof. constructor. by rewrite (set_unfold P P'), (set_unfold Q Q'). Qed.
Lemma set_unfold_iff P Q P' Q' :
  SetUnfold P P' → SetUnfold Q Q' → SetUnfold (P ↔ Q) (P' ↔ Q').
Proof. constructor. by rewrite (set_unfold P P'), (set_unfold Q Q'). Qed.
Lemma set_unfold_not P P' : SetUnfold P P' → SetUnfold (¬P) (¬P').
Proof. constructor. by rewrite (set_unfold P P'). Qed.
Lemma set_unfold_forall {A} (P P' : A → Prop) :
  (∀ x, SetUnfold (P x) (P' x)) → SetUnfold (∀ x, P x) (∀ x, P' x).
Proof. constructor. naive_solver. Qed.
Lemma set_unfold_exist {A} (P P' : A → Prop) :
  (∀ x, SetUnfold (P x) (P' x)) → SetUnfold (∃ x, P x) (∃ x, P' x).
Proof. constructor. naive_solver. Qed.

(* Avoid too eager application of the above instances (and thus too eager
unfolding of type class transparent definitions). *)
Global Hint Extern 0 (SetUnfold (_ → _) _) =>
  class_apply set_unfold_impl : typeclass_instances.
Global Hint Extern 0 (SetUnfold (_ ∧ _) _) =>
  class_apply set_unfold_and : typeclass_instances.
Global Hint Extern 0 (SetUnfold (_ ∨ _) _) =>
  class_apply set_unfold_or : typeclass_instances.
Global Hint Extern 0 (SetUnfold (_ ↔ _) _) =>
  class_apply set_unfold_iff : typeclass_instances.
Global Hint Extern 0 (SetUnfold (¬ _) _) =>
  class_apply set_unfold_not : typeclass_instances.
Global Hint Extern 1 (SetUnfold (∀ _, _) _) =>
  class_apply set_unfold_forall : typeclass_instances.
Global Hint Extern 0 (SetUnfold (∃ _, _) _) =>
  class_apply set_unfold_exist : typeclass_instances.

Section set_unfold_simple.
  Context `{SemiSet A C}.
  Implicit Types x y : A.
  Implicit Types X Y : C.

  Global Instance set_unfold_empty x : SetUnfoldElemOf x (∅ : C) False.
  Proof. constructor. split; [|done]. apply not_elem_of_empty. Qed.
  Global Instance set_unfold_singleton x y : SetUnfoldElemOf x ({[ y ]} : C) (x = y).
  Proof. constructor; apply elem_of_singleton. Qed.
  Global Instance set_unfold_union x X Y P Q :
    SetUnfoldElemOf x X P → SetUnfoldElemOf x Y Q →
    SetUnfoldElemOf x (X ∪ Y) (P ∨ Q).
  Proof.
    intros ??; constructor.
    by rewrite elem_of_union,
      (set_unfold_elem_of x X P), (set_unfold_elem_of x Y Q).
  Qed.
  Global Instance set_unfold_equiv_same X : SetUnfold (X ≡ X) True | 1.
  Proof. done. Qed.
  Global Instance set_unfold_equiv_empty_l X (P : A → Prop) :
    (∀ x, SetUnfoldElemOf x X (P x)) → SetUnfold (∅ ≡ X) (∀ x, ¬P x) | 5.
  Proof.
    intros ?; constructor. unfold equiv, set_equiv.
    pose proof (not_elem_of_empty (C:=C)); naive_solver.
  Qed.
  Global Instance set_unfold_equiv_empty_r (P : A → Prop) X :
    (∀ x, SetUnfoldElemOf x X (P x)) → SetUnfold (X ≡ ∅) (∀ x, ¬P x) | 5.
  Proof.
    intros ?; constructor. unfold equiv, set_equiv.
    pose proof (not_elem_of_empty (C:=C)); naive_solver.
  Qed.
  Global Instance set_unfold_equiv (P Q : A → Prop) X Y :
    (∀ x, SetUnfoldElemOf x X (P x)) → (∀ x, SetUnfoldElemOf x Y (Q x)) →
    SetUnfold (X ≡ Y) (∀ x, P x ↔ Q x) | 10.
  Proof. constructor. apply forall_proper; naive_solver. Qed.
  Global Instance set_unfold_subseteq (P Q : A → Prop) X Y :
    (∀ x, SetUnfoldElemOf x X (P x)) → (∀ x, SetUnfoldElemOf x Y (Q x)) →
    SetUnfold (X ⊆ Y) (∀ x, P x → Q x).
  Proof. constructor. apply forall_proper; naive_solver. Qed.
  Global Instance set_unfold_subset (P Q : A → Prop) X Y :
    (∀ x, SetUnfoldElemOf x X (P x)) → (∀ x, SetUnfoldElemOf x Y (Q x)) →
    SetUnfold (X ⊂ Y) ((∀ x, P x → Q x) ∧ ¬∀ x, Q x → P x).
  Proof.
    constructor. unfold strict.
    repeat f_equiv; apply forall_proper; naive_solver.
  Qed.
  Global Instance set_unfold_disjoint (P Q : A → Prop) X Y :
    (∀ x, SetUnfoldElemOf x X (P x)) → (∀ x, SetUnfoldElemOf x Y (Q x)) →
    SetUnfold (X ## Y) (∀ x, P x → Q x → False).
  Proof. constructor. unfold disjoint, set_disjoint. naive_solver. Qed.

  Context `{!LeibnizEquiv C}.
  Global Instance set_unfold_equiv_same_L X : SetUnfold (X = X) True | 1.
  Proof. done. Qed.
  Global Instance set_unfold_equiv_empty_l_L X (P : A → Prop) :
    (∀ x, SetUnfoldElemOf x X (P x)) → SetUnfold (∅ = X) (∀ x, ¬P x) | 5.
  Proof. constructor. unfold_leibniz. by apply set_unfold_equiv_empty_l. Qed.
  Global Instance set_unfold_equiv_empty_r_L (P : A → Prop) X :
    (∀ x, SetUnfoldElemOf x X (P x)) → SetUnfold (X = ∅) (∀ x, ¬P x) | 5.
  Proof. constructor. unfold_leibniz. by apply set_unfold_equiv_empty_r. Qed.
  Global Instance set_unfold_equiv_L (P Q : A → Prop) X Y :
    (∀ x, SetUnfoldElemOf x X (P x)) → (∀ x, SetUnfoldElemOf x Y (Q x)) →
    SetUnfold (X = Y) (∀ x, P x ↔ Q x) | 10.
  Proof. constructor. unfold_leibniz. by apply set_unfold_equiv. Qed.
End set_unfold_simple.

Section set_unfold.
  Context `{Set_ A C}.
  Implicit Types x y : A.
  Implicit Types X Y : C.

  Global Instance set_unfold_intersection x X Y P Q :
    SetUnfoldElemOf x X P → SetUnfoldElemOf x Y Q →
    SetUnfoldElemOf x (X ∩ Y) (P ∧ Q).
  Proof.
    intros ??; constructor. rewrite elem_of_intersection.
    by rewrite (set_unfold_elem_of x X P), (set_unfold_elem_of x Y Q).
  Qed.
  Global Instance set_unfold_difference x X Y P Q :
    SetUnfoldElemOf x X P → SetUnfoldElemOf x Y Q →
    SetUnfoldElemOf x (X ∖ Y) (P ∧ ¬Q).
  Proof.
    intros ??; constructor. rewrite elem_of_difference.
    by rewrite (set_unfold_elem_of x X P), (set_unfold_elem_of x Y Q).
  Qed.
End set_unfold.

Instance set_unfold_top `{TopSet A C} (x : A) :
  SetUnfoldElemOf x (⊤ : C) True.
Proof. constructor. split; [done|intros; apply elem_of_top']. Qed.

Section set_unfold_monad.
  Context `{MonadSet M}.

  Global Instance set_unfold_ret {A} (x y : A) :
    SetUnfoldElemOf x (mret (M:=M) y) (x = y).
  Proof. constructor; apply elem_of_ret. Qed.
  Global Instance set_unfold_bind {A B} (f : A → M B) X (P Q : A → Prop) x :
    (∀ y, SetUnfoldElemOf y X (P y)) → (∀ y, SetUnfoldElemOf x (f y) (Q y)) →
    SetUnfoldElemOf x (X ≫= f) (∃ y, Q y ∧ P y).
  Proof. constructor. rewrite elem_of_bind; naive_solver. Qed.
  Global Instance set_unfold_fmap {A B} (f : A → B) (X : M A) (P : A → Prop) x :
    (∀ y, SetUnfoldElemOf y X (P y)) →
    SetUnfoldElemOf x (f <$> X) (∃ y, x = f y ∧ P y).
  Proof. constructor. rewrite elem_of_fmap; naive_solver. Qed.
  Global Instance set_unfold_join {A} (X : M (M A)) (P : M A → Prop) x :
    (∀ Y, SetUnfoldElemOf Y X (P Y)) →
    SetUnfoldElemOf x (mjoin X) (∃ Y, x ∈ Y ∧ P Y).
  Proof. constructor. rewrite elem_of_join; naive_solver. Qed.
End set_unfold_monad.

Section set_unfold_list.
  Context {A : Type}.
  Implicit Types x : A.
  Implicit Types l : list A.

  Global Instance set_unfold_nil x : SetUnfoldElemOf x [] False.
  Proof. constructor; apply elem_of_nil. Qed.
  Global Instance set_unfold_cons x y l P :
    SetUnfoldElemOf x l P → SetUnfoldElemOf x (y :: l) (x = y ∨ P).
  Proof. constructor. by rewrite elem_of_cons, (set_unfold_elem_of x l P). Qed.
  Global Instance set_unfold_app x l k P Q :
    SetUnfoldElemOf x l P → SetUnfoldElemOf x k Q →
    SetUnfoldElemOf x (l ++ k) (P ∨ Q).
  Proof.
    intros ??; constructor.
    by rewrite elem_of_app, (set_unfold_elem_of x l P), (set_unfold_elem_of x k Q).
  Qed.
  Global Instance set_unfold_included l k (P Q : A → Prop) :
    (∀ x, SetUnfoldElemOf x l (P x)) → (∀ x, SetUnfoldElemOf x k (Q x)) →
    SetUnfold (l ⊆ k) (∀ x, P x → Q x).
  Proof.
    constructor; unfold subseteq, list_subseteq.
    apply forall_proper; naive_solver.
  Qed.

  Global Instance set_unfold_reverse x l P :
    SetUnfoldElemOf x l P → SetUnfoldElemOf x (reverse l) P.
  Proof. constructor. by rewrite elem_of_reverse, (set_unfold_elem_of x l P). Qed.

  Global Instance set_unfold_list_fmap {B} (f : A → B) l P y :
    (∀ x, SetUnfoldElemOf x l (P x)) →
    SetUnfoldElemOf y (f <$> l) (∃ x, y = f x ∧ P x).
  Proof.
    constructor. rewrite elem_of_list_fmap. f_equiv; intros x.
    by rewrite (set_unfold_elem_of x l (P x)).
  Qed.
  Global Instance set_unfold_rotate x l P n:
    SetUnfoldElemOf x l P → SetUnfoldElemOf x (rotate n l) P.
  Proof. constructor. by rewrite elem_of_rotate, (set_unfold_elem_of x l P). Qed.

End set_unfold_list.

Tactic Notation "set_unfold" :=
  let rec unfold_hyps :=
    try match goal with
    | H : ?P |- _ =>
       lazymatch type of P with
       | Prop =>
         apply set_unfold_1 in H; revert H;
         first [unfold_hyps; intros H | intros H; fail 1]
       | _ => fail
       end
    end in
  apply set_unfold_2; unfold_hyps; csimpl in *.

Tactic Notation "set_unfold" "in" ident(H) :=
  let P := type of H in
  lazymatch type of P with
  | Prop => apply set_unfold_1 in H
  | _ => fail "hypothesis" H "is not a proposition"
  end.

(** Since [firstorder] already fails or loops on very small goals generated by
[set_solver], we use the [naive_solver] tactic as a substitute. *)
Tactic Notation "set_solver" "by" tactic3(tac) :=
  try fast_done;
  intros; setoid_subst;
  set_unfold;
  intros; setoid_subst;
  try match goal with |- _ ∈ _ => apply dec_stable end;
  naive_solver tac.
Tactic Notation "set_solver" "-" hyp_list(Hs) "by" tactic3(tac) :=
  clear Hs; set_solver by tac.
Tactic Notation "set_solver" "+" hyp_list(Hs) "by" tactic3(tac) :=
  clear -Hs; set_solver by tac.
Tactic Notation "set_solver" := set_solver by idtac.
Tactic Notation "set_solver" "-" hyp_list(Hs) := clear Hs; set_solver.
Tactic Notation "set_solver" "+" hyp_list(Hs) := clear -Hs; set_solver.

Global Hint Extern 1000 (_ ∉ _) => set_solver : set_solver.
Global Hint Extern 1000 (_ ∈ _) => set_solver : set_solver.
Global Hint Extern 1000 (_ ⊆ _) => set_solver : set_solver.

(** * Sets with [∪], [∅] and [{[_]}] *)
Section semi_set.
  Context `{SemiSet A C}.
  Implicit Types x y : A.
  Implicit Types X Y : C.
  Implicit Types Xs Ys : list C.

  (** Equality *)
  Lemma elem_of_equiv X Y : X ≡ Y ↔ ∀ x, x ∈ X ↔ x ∈ Y.
  Proof. set_solver. Qed.
  Lemma set_equiv_spec X Y : X ≡ Y ↔ X ⊆ Y ∧ Y ⊆ X.
  Proof. set_solver. Qed.

  (** Subset relation *)
  Global Instance set_subseteq_antisymm: AntiSymm (≡) (⊆@{C}).
  Proof. intros ??. set_solver. Qed.

  Global Instance set_subseteq_preorder: PreOrder (⊆@{C}).
  Proof. split; [by intros ??|]. intros ???; set_solver. Qed.

  Lemma subseteq_union X Y : X ⊆ Y ↔ X ∪ Y ≡ Y.
  Proof. set_solver. Qed.
  Lemma subseteq_union_1 X Y : X ⊆ Y → X ∪ Y ≡ Y.
  Proof. by rewrite subseteq_union. Qed.
  Lemma subseteq_union_2 X Y : X ∪ Y ≡ Y → X ⊆ Y.
  Proof. by rewrite subseteq_union. Qed.

  Lemma union_subseteq_l X Y : X ⊆ X ∪ Y.
  Proof. set_solver. Qed.
  Lemma union_subseteq_r X Y : Y ⊆ X ∪ Y.
  Proof. set_solver. Qed.
  Lemma union_least X Y Z : X ⊆ Z → Y ⊆ Z → X ∪ Y ⊆ Z.
  Proof. set_solver. Qed.

  Lemma elem_of_subseteq X Y : X ⊆ Y ↔ ∀ x, x ∈ X → x ∈ Y.
  Proof. done. Qed.
  Lemma elem_of_subset X Y : X ⊂ Y ↔ (∀ x, x ∈ X → x ∈ Y) ∧ ¬(∀ x, x ∈ Y → x ∈ X).
  Proof. set_solver. Qed.

  (** Union *)
  Lemma union_subseteq X Y Z : X ∪ Y ⊆ Z ↔ X ⊆ Z ∧ Y ⊆ Z.
  Proof. set_solver. Qed.
  Lemma not_elem_of_union x X Y : x ∉ X ∪ Y ↔ x ∉ X ∧ x ∉ Y.
  Proof. set_solver. Qed.
  Lemma elem_of_union_l x X Y : x ∈ X → x ∈ X ∪ Y.
  Proof. set_solver. Qed.
  Lemma elem_of_union_r x X Y : x ∈ Y → x ∈ X ∪ Y.
  Proof. set_solver. Qed.
  Lemma union_mono_l X Y1 Y2 : Y1 ⊆ Y2 → X ∪ Y1 ⊆ X ∪ Y2.
  Proof. set_solver. Qed.
  Lemma union_mono_r X1 X2 Y : X1 ⊆ X2 → X1 ∪ Y ⊆ X2 ∪ Y.
  Proof. set_solver. Qed.
  Lemma union_mono X1 X2 Y1 Y2 : X1 ⊆ X2 → Y1 ⊆ Y2 → X1 ∪ Y1 ⊆ X2 ∪ Y2.
  Proof. set_solver. Qed.

  Global Instance union_idemp : IdemP (≡@{C}) (∪).
  Proof. intros X. set_solver. Qed.
  Global Instance union_empty_l : LeftId (≡@{C}) ∅ (∪).
  Proof. intros X. set_solver. Qed.
  Global Instance union_empty_r : RightId (≡@{C}) ∅ (∪).
  Proof. intros X. set_solver. Qed.
  Global Instance union_comm : Comm (≡@{C}) (∪).
  Proof. intros X Y. set_solver. Qed.
  Global Instance union_assoc : Assoc (≡@{C}) (∪).
  Proof. intros X Y Z. set_solver. Qed.

  Lemma empty_union X Y : X ∪ Y ≡ ∅ ↔ X ≡ ∅ ∧ Y ≡ ∅.
  Proof. set_solver. Qed.

  Lemma union_cancel_l X Y Z : Z ## X → Z ## Y → Z ∪ X ≡ Z ∪ Y → X ≡ Y.
  Proof. set_solver. Qed.
  Lemma union_cancel_r X Y Z : X ## Z → Y ## Z → X ∪ Z ≡ Y ∪ Z → X ≡ Y.
  Proof. set_solver. Qed.

  (** Empty *)
  Lemma empty_subseteq X : ∅ ⊆ X.
  Proof. set_solver. Qed.
  Lemma elem_of_equiv_empty X : X ≡ ∅ ↔ ∀ x, x ∉ X.
  Proof. set_solver. Qed.
  Lemma elem_of_empty x : x ∈ (∅ : C) ↔ False.
  Proof. set_solver. Qed.
  Lemma equiv_empty X : X ⊆ ∅ → X ≡ ∅.
  Proof. set_solver. Qed.
  Lemma union_positive_l X Y : X ∪ Y ≡ ∅ → X ≡ ∅.
  Proof. set_solver. Qed.
  Lemma union_positive_l_alt X Y : X ≢ ∅ → X ∪ Y ≢ ∅.
  Proof. set_solver. Qed.
  Lemma non_empty_inhabited x X : x ∈ X → X ≢ ∅.
  Proof. set_solver. Qed.

  (** Singleton *)
  Lemma elem_of_singleton_1 x y : x ∈ ({[y]} : C) → x = y.
  Proof. by rewrite elem_of_singleton. Qed.
  Lemma elem_of_singleton_2 x y : x = y → x ∈ ({[y]} : C).
  Proof. by rewrite elem_of_singleton. Qed.
  Lemma elem_of_subseteq_singleton x X : x ∈ X ↔ {[ x ]} ⊆ X.
  Proof. set_solver. Qed.
  Lemma non_empty_singleton x : ({[ x ]} : C) ≢ ∅.
  Proof. set_solver. Qed.
  Lemma not_elem_of_singleton x y : x ∉ ({[ y ]} : C) ↔ x ≠ y.
  Proof. by rewrite elem_of_singleton. Qed.
  Lemma not_elem_of_singleton_1 x y : x ∉ ({[ y ]} : C) → x ≠ y.
  Proof. apply not_elem_of_singleton. Qed.
  Lemma not_elem_of_singleton_2 x y : x ≠ y → x ∉ ({[ y ]} : C).
  Proof. apply not_elem_of_singleton. Qed.

  (** Disjointness *)
  Lemma elem_of_disjoint X Y : X ## Y ↔ ∀ x, x ∈ X → x ∈ Y → False.
  Proof. done. Qed.

  Global Instance disjoint_sym : Symmetric (##@{C}).
  Proof. intros X Y. set_solver. Qed.
  Lemma disjoint_empty_l Y : ∅ ## Y.
  Proof. set_solver. Qed.
  Lemma disjoint_empty_r X : X ## ∅.
  Proof. set_solver. Qed.
  Lemma disjoint_singleton_l x Y : {[ x ]} ## Y ↔ x ∉ Y.
  Proof. set_solver. Qed.
  Lemma disjoint_singleton_r y X : X ## {[ y ]} ↔ y ∉ X.
  Proof. set_solver. Qed.
  Lemma disjoint_union_l X1 X2 Y : X1 ∪ X2 ## Y ↔ X1 ## Y ∧ X2 ## Y.
  Proof. set_solver. Qed.
  Lemma disjoint_union_r X Y1 Y2 : X ## Y1 ∪ Y2 ↔ X ## Y1 ∧ X ## Y2.
  Proof. set_solver. Qed.

  (** Big unions *)
  Lemma elem_of_union_list Xs x : x ∈ ⋃ Xs ↔ ∃ X, X ∈ Xs ∧ x ∈ X.
  Proof.
    split.
    - induction Xs; simpl; intros HXs; [by apply elem_of_empty in HXs|].
      setoid_rewrite elem_of_cons. apply elem_of_union in HXs. naive_solver.
    - intros [X [Hx]]. induction Hx; simpl; [by apply elem_of_union_l |].
      intros. apply elem_of_union_r; auto.
  Qed.

  Lemma union_list_nil : ⋃ @nil C = ∅.
  Proof. done. Qed.
  Lemma union_list_cons X Xs : ⋃ (X :: Xs) = X ∪ ⋃ Xs.
  Proof. done. Qed.
  Lemma union_list_singleton X : ⋃ [X] ≡ X.
  Proof. simpl. by rewrite (right_id ∅ _). Qed.
  Lemma union_list_app Xs1 Xs2 : ⋃ (Xs1 ++ Xs2) ≡ ⋃ Xs1 ∪ ⋃ Xs2.
  Proof.
    induction Xs1 as [|X Xs1 IH]; simpl; [by rewrite (left_id ∅ _)|].
    by rewrite IH, (assoc _).
  Qed.
  Lemma union_list_reverse Xs : ⋃ (reverse Xs) ≡ ⋃ Xs.
  Proof.
    induction Xs as [|X Xs IH]; simpl; [done |].
    by rewrite reverse_cons, union_list_app,
      union_list_singleton, (comm _), IH.
  Qed.
  Lemma union_list_mono Xs Ys : Xs ⊆* Ys → ⋃ Xs ⊆ ⋃ Ys.
  Proof. induction 1; simpl; auto using union_mono. Qed.
  Lemma empty_union_list Xs : ⋃ Xs ≡ ∅ ↔ Forall (.≡ ∅) Xs.
  Proof.
    split.
    - induction Xs; simpl; rewrite ?empty_union; intuition.
    - induction 1 as [|?? E1 ? E2]; simpl; [done|]. by apply empty_union.
  Qed.

  Section leibniz.
    Context `{!LeibnizEquiv C}.

    Lemma elem_of_equiv_L X Y : X = Y ↔ ∀ x, x ∈ X ↔ x ∈ Y.
    Proof. unfold_leibniz. apply elem_of_equiv. Qed.
    Lemma set_equiv_spec_L X Y : X = Y ↔ X ⊆ Y ∧ Y ⊆ X.
    Proof. unfold_leibniz. apply set_equiv_spec. Qed.

    (** Subset relation *)
    Global Instance set_subseteq_partialorder : PartialOrder (⊆@{C}).
    Proof. split; [apply _|]. intros ??. unfold_leibniz. apply (anti_symm _). Qed.

    Lemma subseteq_union_L X Y : X ⊆ Y ↔ X ∪ Y = Y.
    Proof. unfold_leibniz. apply subseteq_union. Qed.
    Lemma subseteq_union_1_L X Y : X ⊆ Y → X ∪ Y = Y.
    Proof. unfold_leibniz. apply subseteq_union_1. Qed.
    Lemma subseteq_union_2_L X Y : X ∪ Y = Y → X ⊆ Y.
    Proof. unfold_leibniz. apply subseteq_union_2. Qed.

    (** Union *)
    Global Instance union_idemp_L : IdemP (=@{C}) (∪).
    Proof. intros ?. unfold_leibniz. apply (idemp _). Qed.
    Global Instance union_empty_l_L : LeftId (=@{C}) ∅ (∪).
    Proof. intros ?. unfold_leibniz. apply (left_id _ _). Qed.
    Global Instance union_empty_r_L : RightId (=@{C}) ∅ (∪).
    Proof. intros ?. unfold_leibniz. apply (right_id _ _). Qed.
    Global Instance union_comm_L : Comm (=@{C}) (∪).
    Proof. intros ??. unfold_leibniz. apply (comm _). Qed.
    Global Instance union_assoc_L : Assoc (=@{C}) (∪).
    Proof. intros ???. unfold_leibniz. apply (assoc _). Qed.

    Lemma empty_union_L X Y : X ∪ Y = ∅ ↔ X = ∅ ∧ Y = ∅.
    Proof. unfold_leibniz. apply empty_union. Qed.

    Lemma union_cancel_l_L X Y Z : Z ## X → Z ## Y → Z ∪ X = Z ∪ Y → X = Y.
    Proof. unfold_leibniz. apply union_cancel_l. Qed.
    Lemma union_cancel_r_L X Y Z : X ## Z → Y ## Z → X ∪ Z = Y ∪ Z → X = Y.
    Proof. unfold_leibniz. apply union_cancel_r. Qed.

    (** Empty *)
    Lemma elem_of_equiv_empty_L X : X = ∅ ↔ ∀ x, x ∉ X.
    Proof. unfold_leibniz. apply elem_of_equiv_empty. Qed.
    Lemma equiv_empty_L X : X ⊆ ∅ → X = ∅.
    Proof. unfold_leibniz. apply equiv_empty. Qed.
    Lemma union_positive_l_L X Y : X ∪ Y = ∅ → X = ∅.
    Proof. unfold_leibniz. apply union_positive_l. Qed.
    Lemma union_positive_l_alt_L X Y : X ≠ ∅ → X ∪ Y ≠ ∅.
    Proof. unfold_leibniz. apply union_positive_l_alt. Qed.
    Lemma non_empty_inhabited_L x X : x ∈ X → X ≠ ∅.
    Proof. unfold_leibniz. apply non_empty_inhabited. Qed.

    (** Singleton *)
    Lemma non_empty_singleton_L x : {[ x ]} ≠ (∅ : C).
    Proof. unfold_leibniz. apply non_empty_singleton. Qed.

    (** Big unions *)
    Lemma union_list_singleton_L X : ⋃ [X] = X.
    Proof. unfold_leibniz. apply union_list_singleton. Qed.
    Lemma union_list_app_L Xs1 Xs2 : ⋃ (Xs1 ++ Xs2) = ⋃ Xs1 ∪ ⋃ Xs2.
    Proof. unfold_leibniz. apply union_list_app. Qed.
    Lemma union_list_reverse_L Xs : ⋃ (reverse Xs) = ⋃ Xs.
    Proof. unfold_leibniz. apply union_list_reverse. Qed.
    Lemma empty_union_list_L Xs : ⋃ Xs = ∅ ↔ Forall (.= ∅) Xs.
    Proof. unfold_leibniz. by rewrite empty_union_list. Qed.
  End leibniz.

  Lemma not_elem_of_iff `{!RelDecision (∈@{C})} X Y x :
    (x ∈ X ↔ x ∈ Y) ↔ (x ∉ X ↔ x ∉ Y).
  Proof. destruct (decide (x ∈ X)), (decide (x ∈ Y)); tauto. Qed.

  Section dec.
    Context `{!RelDecision (≡@{C})}.
    Lemma set_subseteq_inv X Y : X ⊆ Y → X ⊂ Y ∨ X ≡ Y.
    Proof. destruct (decide (X ≡ Y)); [by right|left;set_solver]. Qed.
    Lemma set_not_subset_inv X Y : X ⊄ Y → X ⊈ Y ∨ X ≡ Y.
    Proof. destruct (decide (X ≡ Y)); [by right|left;set_solver]. Qed.

    Lemma non_empty_union X Y : X ∪ Y ≢ ∅ ↔ X ≢ ∅ ∨ Y ≢ ∅.
    Proof. rewrite empty_union. destruct (decide (X ≡ ∅)); intuition. Qed.
    Lemma non_empty_union_list Xs : ⋃ Xs ≢ ∅ → Exists (.≢ ∅) Xs.
    Proof. rewrite empty_union_list. apply (not_Forall_Exists _). Qed.

    Context `{!LeibnizEquiv C}.
    Lemma set_subseteq_inv_L X Y : X ⊆ Y → X ⊂ Y ∨ X = Y.
    Proof. unfold_leibniz. apply set_subseteq_inv. Qed.
    Lemma set_not_subset_inv_L X Y : X ⊄ Y → X ⊈ Y ∨ X = Y.
    Proof. unfold_leibniz. apply set_not_subset_inv. Qed.
    Lemma non_empty_union_L X Y : X ∪ Y ≠ ∅ ↔ X ≠ ∅ ∨ Y ≠ ∅.
    Proof. unfold_leibniz. apply non_empty_union. Qed.
    Lemma non_empty_union_list_L Xs : ⋃ Xs ≠ ∅ → Exists (.≠ ∅) Xs.
    Proof. unfold_leibniz. apply non_empty_union_list. Qed.
  End dec.
End semi_set.

(** * Sets with [∪], [∩], [∖], [∅] and [{[_]}] *)
Section set.
  Context `{Set_ A C}.
  Implicit Types x y : A.
  Implicit Types X Y : C.

  (** Intersection *)
  Lemma subseteq_intersection X Y : X ⊆ Y ↔ X ∩ Y ≡ X.
  Proof. set_solver. Qed.
  Lemma subseteq_intersection_1 X Y : X ⊆ Y → X ∩ Y ≡ X.
  Proof. apply subseteq_intersection. Qed.
  Lemma subseteq_intersection_2 X Y : X ∩ Y ≡ X → X ⊆ Y.
  Proof. apply subseteq_intersection. Qed.

  Lemma intersection_subseteq_l X Y : X ∩ Y ⊆ X.
  Proof. set_solver. Qed.
  Lemma intersection_subseteq_r X Y : X ∩ Y ⊆ Y.
  Proof. set_solver. Qed.
  Lemma intersection_greatest X Y Z : Z ⊆ X → Z ⊆ Y → Z ⊆ X ∩ Y.
  Proof. set_solver. Qed.

  Lemma intersection_mono_l X Y1 Y2 : Y1 ⊆ Y2 → X ∩ Y1 ⊆ X ∩ Y2.
  Proof. set_solver. Qed.
  Lemma intersection_mono_r X1 X2 Y : X1 ⊆ X2 → X1 ∩ Y ⊆ X2 ∩ Y.
  Proof. set_solver. Qed.
  Lemma intersection_mono X1 X2 Y1 Y2 :
    X1 ⊆ X2 → Y1 ⊆ Y2 → X1 ∩ Y1 ⊆ X2 ∩ Y2.
  Proof. set_solver. Qed.

  Global Instance intersection_idemp : IdemP (≡@{C}) (∩).
  Proof. intros X; set_solver. Qed.
  Global Instance intersection_comm : Comm (≡@{C}) (∩).
  Proof. intros X Y; set_solver. Qed.
  Global Instance intersection_assoc : Assoc (≡@{C}) (∩).
  Proof. intros X Y Z; set_solver. Qed.
  Global Instance intersection_empty_l : LeftAbsorb (≡@{C}) ∅ (∩).
  Proof. intros X; set_solver. Qed.
  Global Instance intersection_empty_r: RightAbsorb (≡@{C}) ∅ (∩).
  Proof. intros X; set_solver. Qed.

  Lemma intersection_singletons x : ({[x]} : C) ∩ {[x]} ≡ {[x]}.
  Proof. set_solver. Qed.

  Lemma union_intersection_l X Y Z : X ∪ (Y ∩ Z) ≡ (X ∪ Y) ∩ (X ∪ Z).
  Proof. set_solver. Qed.
  Lemma union_intersection_r X Y Z : (X ∩ Y) ∪ Z ≡ (X ∪ Z) ∩ (Y ∪ Z).
  Proof. set_solver. Qed.
  Lemma intersection_union_l X Y Z : X ∩ (Y ∪ Z) ≡ (X ∩ Y) ∪ (X ∩ Z).
  Proof. set_solver. Qed.
  Lemma intersection_union_r X Y Z : (X ∪ Y) ∩ Z ≡ (X ∩ Z) ∪ (Y ∩ Z).
  Proof. set_solver. Qed.

  (** Difference *)
  Lemma difference_twice X Y : (X ∖ Y) ∖ Y ≡ X ∖ Y.
  Proof. set_solver. Qed.
  Lemma subseteq_empty_difference X Y : X ⊆ Y → X ∖ Y ≡ ∅.
  Proof. set_solver. Qed.
  Lemma difference_diag X : X ∖ X ≡ ∅.
  Proof. set_solver. Qed.
  Lemma difference_empty X : X ∖ ∅ ≡ X.
  Proof. set_solver. Qed.
  Lemma difference_union_distr_l X Y Z : (X ∪ Y) ∖ Z ≡ X ∖ Z ∪ Y ∖ Z.
  Proof. set_solver. Qed.
  Lemma difference_union_distr_r X Y Z : Z ∖ (X ∪ Y) ≡ (Z ∖ X) ∩ (Z ∖ Y).
  Proof. set_solver. Qed.
  Lemma difference_intersection_distr_l X Y Z : (X ∩ Y) ∖ Z ≡ X ∖ Z ∩ Y ∖ Z.
  Proof. set_solver. Qed.
  Lemma difference_disjoint X Y : X ## Y → X ∖ Y ≡ X.
  Proof. set_solver. Qed.
  Lemma subset_difference_elem_of {x: A} {s: C} (inx: x ∈ s): s ∖ {[ x ]} ⊂ s.
  Proof. set_solver. Qed.
  Lemma difference_difference X Y Z : (X ∖ Y) ∖ Z ≡ X ∖ (Y ∪ Z).
  Proof. set_solver. Qed.

  Lemma difference_mono X1 X2 Y1 Y2 :
    X1 ⊆ X2 → Y2 ⊆ Y1 → X1 ∖ Y1 ⊆ X2 ∖ Y2.
  Proof. set_solver. Qed.
  Lemma difference_mono_l X Y1 Y2 : Y2 ⊆ Y1 → X ∖ Y1 ⊆ X ∖ Y2.
  Proof. set_solver. Qed.
  Lemma difference_mono_r X1 X2 Y : X1 ⊆ X2 → X1 ∖ Y ⊆ X2 ∖ Y.
  Proof. set_solver. Qed.

  Lemma subseteq_difference_r X Y1 Y2 :
    X ## Y2 → X ⊆ Y1 → X ⊆ Y1 ∖ Y2.
  Proof. set_solver. Qed.
  Lemma subseteq_difference_l X1 X2 Y : X1 ⊆ Y → X1 ∖ X2 ⊆ Y.
  Proof. set_solver. Qed.

  (** Disjointness *)
  Lemma disjoint_intersection X Y : X ## Y ↔ X ∩ Y ≡ ∅.
  Proof. set_solver. Qed.
  Lemma disjoint_difference_l1 X1 X2 Y : Y ⊆ X2 → X1 ∖ X2 ## Y.
  Proof. set_solver. Qed.
  Lemma disjoint_difference_l2 X1 X2 Y : X1 ## Y → X1 ∖ X2 ## Y.
  Proof. set_solver. Qed.
  Lemma disjoint_difference_r1 X Y1 Y2 : X ⊆ Y2 → X ## Y1 ∖ Y2.
  Proof. set_solver. Qed.
  Lemma disjoint_difference_r2 X Y1 Y2 : X ## Y1 → X ## Y1 ∖ Y2.
  Proof. set_solver. Qed.

  Section leibniz.
    Context `{!LeibnizEquiv C}.

    (** Intersection *)
    Lemma subseteq_intersection_L X Y : X ⊆ Y ↔ X ∩ Y = X.
    Proof. unfold_leibniz. apply subseteq_intersection. Qed.
    Lemma subseteq_intersection_1_L X Y : X ⊆ Y → X ∩ Y = X.
    Proof. unfold_leibniz. apply subseteq_intersection_1. Qed.
    Lemma subseteq_intersection_2_L X Y : X ∩ Y = X → X ⊆ Y.
    Proof. unfold_leibniz. apply subseteq_intersection_2. Qed.

    Global Instance intersection_idemp_L : IdemP (=@{C}) (∩).
    Proof. intros ?. unfold_leibniz. apply (idemp _). Qed.
    Global Instance intersection_comm_L : Comm (=@{C}) (∩).
    Proof. intros ??. unfold_leibniz. apply (comm _). Qed.
    Global Instance intersection_assoc_L : Assoc (=@{C}) (∩).
    Proof. intros ???. unfold_leibniz. apply (assoc _). Qed.
    Global Instance intersection_empty_l_L: LeftAbsorb (=@{C}) ∅ (∩).
    Proof. intros ?. unfold_leibniz. apply (left_absorb _ _). Qed.
    Global Instance intersection_empty_r_L: RightAbsorb (=@{C}) ∅ (∩).
    Proof. intros ?. unfold_leibniz. apply (right_absorb _ _). Qed.

    Lemma intersection_singletons_L x : {[x]} ∩ {[x]} = ({[x]} : C).
    Proof. unfold_leibniz. apply intersection_singletons. Qed.

    Lemma union_intersection_l_L X Y Z : X ∪ (Y ∩ Z) = (X ∪ Y) ∩ (X ∪ Z).
    Proof. unfold_leibniz; apply union_intersection_l. Qed.
    Lemma union_intersection_r_L X Y Z : (X ∩ Y) ∪ Z = (X ∪ Z) ∩ (Y ∪ Z).
    Proof. unfold_leibniz; apply union_intersection_r. Qed.
    Lemma intersection_union_l_L X Y Z : X ∩ (Y ∪ Z) = (X ∩ Y) ∪ (X ∩ Z).
    Proof. unfold_leibniz; apply intersection_union_l. Qed.
    Lemma intersection_union_r_L X Y Z : (X ∪ Y) ∩ Z = (X ∩ Z) ∪ (Y ∩ Z).
    Proof. unfold_leibniz; apply intersection_union_r. Qed.

    (** Difference *)
    Lemma difference_twice_L X Y : (X ∖ Y) ∖ Y = X ∖ Y.
    Proof. unfold_leibniz. apply difference_twice. Qed.
    Lemma subseteq_empty_difference_L X Y : X ⊆ Y → X ∖ Y = ∅.
    Proof. unfold_leibniz. apply subseteq_empty_difference. Qed.
    Lemma difference_diag_L X : X ∖ X = ∅.
    Proof. unfold_leibniz. apply difference_diag. Qed.
    Lemma difference_empty_L X : X ∖ ∅ = X.
    Proof. unfold_leibniz. apply difference_empty. Qed.
    Lemma difference_union_distr_l_L X Y Z : (X ∪ Y) ∖ Z = X ∖ Z ∪ Y ∖ Z.
    Proof. unfold_leibniz. apply difference_union_distr_l. Qed.
    Lemma difference_union_distr_r_L X Y Z : Z ∖ (X ∪ Y) = (Z ∖ X) ∩ (Z ∖ Y).
    Proof. unfold_leibniz. apply difference_union_distr_r. Qed.
    Lemma difference_intersection_distr_l_L X Y Z :
      (X ∩ Y) ∖ Z = X ∖ Z ∩ Y ∖ Z.
    Proof. unfold_leibniz. apply difference_intersection_distr_l. Qed.
    Lemma difference_disjoint_L X Y : X ## Y → X ∖ Y = X.
    Proof. unfold_leibniz. apply difference_disjoint. Qed.
    Lemma difference_difference_L X Y Z : (X ∖ Y) ∖ Z = X ∖ (Y ∪ Z).
    Proof. unfold_leibniz. apply difference_difference. Qed.

    (** Disjointness *)
    Lemma disjoint_intersection_L X Y : X ## Y ↔ X ∩ Y = ∅.
    Proof. unfold_leibniz. apply disjoint_intersection. Qed.
  End leibniz.

  Section dec.
    Context `{!RelDecision (∈@{C})}.
    Lemma not_elem_of_intersection x X Y : x ∉ X ∩ Y ↔ x ∉ X ∨ x ∉ Y.
    Proof. rewrite elem_of_intersection. destruct (decide (x ∈ X)); tauto. Qed.
    Lemma not_elem_of_difference x X Y : x ∉ X ∖ Y ↔ x ∉ X ∨ x ∈ Y.
    Proof. rewrite elem_of_difference. destruct (decide (x ∈ Y)); tauto. Qed.
    Lemma union_difference X Y : X ⊆ Y → Y ≡ X ∪ Y ∖ X.
    Proof.
      intros ? x; split; rewrite !elem_of_union, elem_of_difference; [|intuition].
      destruct (decide (x ∈ X)); intuition.
    Qed.
    Lemma difference_union X Y : X ∖ Y ∪ Y ≡ X ∪ Y.
    Proof.
      intros x. rewrite !elem_of_union; rewrite elem_of_difference.
      split; [ | destruct (decide (x ∈ Y)) ]; intuition.
    Qed.
    Lemma subseteq_disjoint_union X Y : X ⊆ Y ↔ ∃ Z, Y ≡ X ∪ Z ∧ X ## Z.
    Proof.
      split; [|set_solver].
      exists (Y ∖ X); split; [auto using union_difference|set_solver].
    Qed.
    Lemma non_empty_difference X Y : X ⊂ Y → Y ∖ X ≢ ∅.
    Proof. intros [HXY1 HXY2] Hdiff. destruct HXY2. set_solver. Qed.
    Lemma empty_difference_subseteq X Y : X ∖ Y ≡ ∅ → X ⊆ Y.
    Proof. set_solver. Qed.
    Lemma singleton_union_difference X Y x :
      {[x]} ∪ (X ∖ Y) ≡ ({[x]} ∪ X) ∖ (Y ∖ {[x]}).
    Proof.
      intro y; split; intros Hy; [ set_solver | ].
      destruct (decide (y ∈ ({[x]} : C))); set_solver.
    Qed.

    Context `{!LeibnizEquiv C}.
    Lemma union_difference_L X Y : X ⊆ Y → Y = X ∪ Y ∖ X.
    Proof. unfold_leibniz. apply union_difference. Qed.
    Lemma difference_union_L X Y : X ∖ Y ∪ Y = X ∪ Y.
    Proof. unfold_leibniz. apply difference_union. Qed.
    Lemma non_empty_difference_L X Y : X ⊂ Y → Y ∖ X ≠ ∅.
    Proof. unfold_leibniz. apply non_empty_difference. Qed.
    Lemma empty_difference_subseteq_L X Y : X ∖ Y = ∅ → X ⊆ Y.
    Proof. unfold_leibniz. apply empty_difference_subseteq. Qed.
    Lemma subseteq_disjoint_union_L X Y : X ⊆ Y ↔ ∃ Z, Y = X ∪ Z ∧ X ## Z.
    Proof. unfold_leibniz. apply subseteq_disjoint_union. Qed.
    Lemma singleton_union_difference_L X Y x :
      {[x]} ∪ (X ∖ Y) = ({[x]} ∪ X) ∖ (Y ∖ {[x]}).
    Proof. unfold_leibniz. apply singleton_union_difference. Qed.
  End dec.
End set.

(** * Sets with [∪], [∩], [∖], [∅], [{[_]}], and [⊤] *)
Section top_set.
  Context `{TopSet A C}.
  Implicit Types x y : A.
  Implicit Types X Y : C.

  Lemma elem_of_top x : x ∈@{C} ⊤ ↔ True.
  Proof. split; [done|intros; apply elem_of_top']. Qed.
  Lemma top_subseteq X : X ⊆ ⊤.
  Proof. intros x. by rewrite elem_of_top. Qed.
End top_set.

(** * Conversion of option and list *)
Section option_and_list_to_set.
  Context `{SemiSet A C}.
  Implicit Types l : list A.

  Lemma elem_of_option_to_set (x : A) mx: x ∈ option_to_set (C:=C) mx ↔ mx = Some x.
  Proof. destruct mx; set_solver. Qed.
  Lemma not_elem_of_option_to_set (x : A) mx: x ∉ option_to_set (C:=C) mx ↔ mx ≠ Some x.
  Proof. by rewrite elem_of_option_to_set. Qed.

  Lemma elem_of_list_to_set (x : A) l : x ∈ list_to_set (C:=C) l ↔ x ∈ l.
  Proof.
    split.
    - induction l; simpl; [by rewrite elem_of_empty|].
      rewrite elem_of_union,elem_of_singleton; intros [->|?]; constructor; auto.
    - induction 1; simpl; rewrite elem_of_union, elem_of_singleton; auto.
  Qed.
  Lemma not_elem_of_list_to_set (x : A) l : x ∉ list_to_set (C:=C) l ↔ x ∉ l.
  Proof. by rewrite elem_of_list_to_set. Qed.

  Global Instance set_unfold_option_to_set (mx : option A) x :
    SetUnfoldElemOf x (option_to_set (C:=C) mx) (mx = Some x).
  Proof. constructor; apply elem_of_option_to_set. Qed.
  Global Instance set_unfold_list_to_set (l : list A) x P :
    SetUnfoldElemOf x l P → SetUnfoldElemOf x (list_to_set (C:=C) l) P.
  Proof. constructor. by rewrite elem_of_list_to_set, (set_unfold (x ∈ l) P). Qed.

  Lemma list_to_set_nil : list_to_set [] =@{C} ∅.
  Proof. done. Qed.
  Lemma list_to_set_cons x l : list_to_set (x :: l) =@{C} {[ x ]} ∪ list_to_set l.
  Proof. done. Qed.
  Lemma list_to_set_app l1 l2 : list_to_set (l1 ++ l2) ≡@{C} list_to_set l1 ∪ list_to_set l2.
  Proof. set_solver. Qed.
  Global Instance list_to_set_perm : Proper ((≡ₚ) ==> (≡)) (list_to_set (C:=C)).
  Proof. induction 1; set_solver. Qed.

  Context `{!LeibnizEquiv C}.
  Lemma list_to_set_app_L l1 l2 : list_to_set (l1 ++ l2) =@{C} list_to_set l1 ∪ list_to_set l2.
  Proof. set_solver. Qed.
  Global Instance list_to_set_perm_L : Proper ((≡ₚ) ==> (=)) (list_to_set (C:=C)).
  Proof. induction 1; set_solver. Qed.
End option_and_list_to_set.

(** * Finite types to sets. *)
Definition fin_to_set (A : Type) `{Singleton A C, Empty C, Union C, Finite A} : C :=
  list_to_set (enum A).

Section fin_to_set.
  Context `{SemiSet A C, Finite A}.
  Implicit Types a : A.

  Lemma elem_of_fin_to_set a : a ∈@{C} fin_to_set A.
  Proof. apply elem_of_list_to_set, elem_of_enum. Qed.

  Global Instance set_unfold_fin_to_set a :
    SetUnfoldElemOf (C:=C) a (fin_to_set A) True.
  Proof. constructor. split; auto using elem_of_fin_to_set. Qed.
End fin_to_set.

(** * Guard *)
Global Instance set_guard `{MonadSet M} : MGuard M :=
  λ P dec A x, match dec with left H => x H | _ => ∅ end.

Section set_monad_base.
  Context `{MonadSet M}.
  Lemma elem_of_guard `{Decision P} {A} (x : A) (X : M A) :
    (x ∈ guard P; X) ↔ P ∧ x ∈ X.
  Proof.
    unfold mguard, set_guard; simpl; case_match;
      rewrite ?elem_of_empty; naive_solver.
  Qed.
  Lemma elem_of_guard_2 `{Decision P} {A} (x : A) (X : M A) :
    P → x ∈ X → x ∈ guard P; X.
  Proof. by rewrite elem_of_guard. Qed.
  Lemma guard_empty `{Decision P} {A} (X : M A) : (guard P; X) ≡ ∅ ↔ ¬P ∨ X ≡ ∅.
  Proof.
    rewrite !elem_of_equiv_empty; setoid_rewrite elem_of_guard.
    destruct (decide P); naive_solver.
  Qed.
  Global Instance set_unfold_guard `{Decision P} {A} (x : A) (X : M A) Q :
    SetUnfoldElemOf x X Q → SetUnfoldElemOf x (guard P; X) (P ∧ Q).
  Proof. constructor. by rewrite elem_of_guard, (set_unfold (x ∈ X) Q). Qed.
  Lemma bind_empty {A B} (f : A → M B) X :
    X ≫= f ≡ ∅ ↔ X ≡ ∅ ∨ ∀ x, x ∈ X → f x ≡ ∅.
  Proof. set_solver. Qed.
End set_monad_base.

(** * Quantifiers *)
Definition set_Forall `{ElemOf A C} (P : A → Prop) (X : C) := ∀ x, x ∈ X → P x.
Definition set_Exists `{ElemOf A C} (P : A → Prop) (X : C) := ∃ x, x ∈ X ∧ P x.

Section quantifiers.
  Context `{SemiSet A C} (P : A → Prop).
  Implicit Types X Y : C.

  Lemma set_Forall_empty : set_Forall P (∅ : C).
  Proof. unfold set_Forall. set_solver. Qed.
  Lemma set_Forall_singleton x : set_Forall P ({[ x ]} : C) ↔ P x.
  Proof. unfold set_Forall. set_solver. Qed.
  Lemma set_Forall_union X Y :
    set_Forall P X → set_Forall P Y → set_Forall P (X ∪ Y).
  Proof. unfold set_Forall. set_solver. Qed.
  Lemma set_Forall_union_inv_1 X Y : set_Forall P (X ∪ Y) → set_Forall P X.
  Proof. unfold set_Forall. set_solver. Qed.
  Lemma set_Forall_union_inv_2 X Y : set_Forall P (X ∪ Y) → set_Forall P Y.
  Proof. unfold set_Forall. set_solver. Qed.
  Lemma set_Forall_list_to_set l : set_Forall P (list_to_set (C:=C) l) ↔ Forall P l.
  Proof. rewrite Forall_forall. unfold set_Forall. set_solver. Qed.

  Lemma set_Exists_empty : ¬set_Exists P (∅ : C).
  Proof. unfold set_Exists. set_solver. Qed.
  Lemma set_Exists_singleton x : set_Exists P ({[ x ]} : C) ↔ P x.
  Proof. unfold set_Exists. set_solver. Qed.
  Lemma set_Exists_union_1 X Y : set_Exists P X → set_Exists P (X ∪ Y).
  Proof. unfold set_Exists. set_solver. Qed.
  Lemma set_Exists_union_2 X Y : set_Exists P Y → set_Exists P (X ∪ Y).
  Proof. unfold set_Exists. set_solver. Qed.
  Lemma set_Exists_union_inv X Y :
    set_Exists P (X ∪ Y) → set_Exists P X ∨ set_Exists P Y.
  Proof. unfold set_Exists. set_solver. Qed.
  Lemma set_Exists_list_to_set l : set_Exists P (list_to_set (C:=C) l) ↔ Exists P l.
  Proof. rewrite Exists_exists. unfold set_Exists. set_solver. Qed.
End quantifiers.

Section more_quantifiers.
  Context `{SemiSet A C}.
  Implicit Types X : C.

  Lemma set_Forall_impl (P Q : A → Prop) X :
    set_Forall P X → (∀ x, P x → Q x) → set_Forall Q X.
  Proof. unfold set_Forall. naive_solver. Qed.
  Lemma set_Exists_impl (P Q : A → Prop) X :
    set_Exists P X → (∀ x, P x → Q x) → set_Exists Q X.
  Proof. unfold set_Exists. naive_solver. Qed.
End more_quantifiers.

(** * Properties of implementations of sets that form a monad *)
Section set_monad.
  Context `{MonadSet M}.

  Global Instance set_fmap_mono {A B} :
    Proper (pointwise_relation _ (=) ==> (⊆) ==> (⊆)) (@fmap M _ A B).
  Proof. intros f g ? X Y ?; set_solver by eauto. Qed.
  Global Instance set_bind_mono {A B} :
    Proper (pointwise_relation _ (⊆) ==> (⊆) ==> (⊆)) (@mbind M _ A B).
  Proof. unfold respectful, pointwise_relation; intros f g Hfg X Y ?. set_solver. Qed.
  Global Instance set_join_mono {A} :
    Proper ((⊆) ==> (⊆)) (@mjoin M _ A).
  Proof. intros X Y ?; set_solver. Qed.

  Lemma set_bind_singleton {A B} (f : A → M B) x : {[ x ]} ≫= f ≡ f x.
  Proof. set_solver. Qed.
  Lemma set_guard_True {A} `{Decision P} (X : M A) : P → (guard P; X) ≡ X.
  Proof. set_solver. Qed.
  Lemma set_fmap_compose {A B C} (f : A → B) (g : B → C) (X : M A) :
    g ∘ f <$> X ≡ g <$> (f <$> X).
  Proof. set_solver. Qed.
  Lemma elem_of_fmap_1 {A B} (f : A → B) (X : M A) (y : B) :
    y ∈ f <$> X → ∃ x, y = f x ∧ x ∈ X.
  Proof. set_solver. Qed.
  Lemma elem_of_fmap_2 {A B} (f : A → B) (X : M A) (x : A) :
    x ∈ X → f x ∈ f <$> X.
  Proof. set_solver. Qed.
  Lemma elem_of_fmap_2_alt {A B} (f : A → B) (X : M A) (x : A) (y : B) :
    x ∈ X → y = f x → y ∈ f <$> X.
  Proof. set_solver. Qed.

  Lemma elem_of_mapM {A B} (f : A → M B) l k :
    l ∈ mapM f k ↔ Forall2 (λ x y, x ∈ f y) l k.
  Proof.
    split.
    - revert l. induction k; set_solver by eauto.
    - induction 1; set_solver.
  Qed.
  Lemma set_mapM_length {A B} (f : A → M B) l k :
    l ∈ mapM f k → length l = length k.
  Proof. revert l; induction k; set_solver by eauto. Qed.
  Lemma elem_of_mapM_fmap {A B} (f : A → B) (g : B → M A) l k :
    Forall (λ x, ∀ y, y ∈ g x → f y = x) l → k ∈ mapM g l → fmap f k = l.
  Proof. intros Hl. revert k. induction Hl; set_solver. Qed.
  Lemma elem_of_mapM_Forall {A B} (f : A → M B) (P : B → Prop) l k :
    l ∈ mapM f k → Forall (λ x, ∀ y, y ∈ f x → P y) k → Forall P l.
  Proof. rewrite elem_of_mapM. apply Forall2_Forall_l. Qed.
  Lemma elem_of_mapM_Forall2_l {A B C} (f : A → M B) (P: B → C → Prop) l1 l2 k :
    l1 ∈ mapM f k → Forall2 (λ x y, ∀ z, z ∈ f x → P z y) k l2 →
    Forall2 P l1 l2.
  Proof.
    rewrite elem_of_mapM. intros Hl1. revert l2.
    induction Hl1; inversion_clear 1; constructor; auto.
  Qed.
End set_monad.

(** Finite sets *)
Definition pred_finite {A} (P : A → Prop) := ∃ xs : list A, ∀ x, P x → x ∈ xs.
Definition set_finite `{ElemOf A B} (X : B) := pred_finite (.∈ X).

Definition pred_infinite {A} (P : A → Prop) := ∀ xs : list A, ∃ x, P x ∧ x ∉ xs.
Definition set_infinite `{ElemOf A C} (X : C) := pred_infinite (.∈ X).

Section pred_finite_infinite.
  Lemma pred_finite_impl {A} (P Q : A → Prop) :
    pred_finite P → (∀ x, Q x → P x) → pred_finite Q.
  Proof. unfold pred_finite. set_solver. Qed.

  Lemma pred_infinite_impl {A} (P Q : A → Prop) :
    pred_infinite P → (∀ x, P x → Q x) → pred_infinite Q.
  Proof. unfold pred_infinite. set_solver. Qed.

  Lemma pred_not_infinite_finite {A} (P : A → Prop) :
    pred_infinite P → pred_finite P → False.
  Proof. intros Hinf [xs ?]. destruct (Hinf xs). set_solver. Qed.

  Lemma pred_infinite_True `{Infinite A} : pred_infinite (λ _: A, True).
  Proof.
    intros xs. exists (fresh xs). split; [done|]. apply infinite_is_fresh.
  Qed.

  Lemma pred_finite_lt n : pred_finite (flip lt n).
  Proof.
    exists (seq 0 n); intros i Hi. apply (elem_of_list_lookup_2 _ i).
    by rewrite lookup_seq.
  Qed.
  Lemma pred_infinite_lt n : pred_infinite (lt n).
  Proof.
    intros l. exists (S (n `max` max_list l)). split; [lia| ].
    intros H%max_list_elem_of_le; lia.
  Qed.

  Lemma pred_finite_le n : pred_finite (flip le n).
  Proof. eapply pred_finite_impl; [apply (pred_finite_lt (S n))|]; naive_solver lia. Qed.
  Lemma pred_infinite_le n : pred_infinite (le n).
  Proof. eapply pred_infinite_impl; [apply (pred_infinite_lt (S n))|]; naive_solver lia. Qed.
End pred_finite_infinite.

Section set_finite_infinite.
  Context `{SemiSet A C}.
  Implicit Types X Y : C.

  Lemma set_not_infinite_finite X : set_infinite X → set_finite X → False.
  Proof. apply pred_not_infinite_finite. Qed.

  Global Instance set_finite_subseteq :
    Proper (flip (⊆) ==> impl) (@set_finite A C _).
  Proof. intros X Y HX ?. eapply pred_finite_impl; set_solver. Qed.
  Global Instance set_finite_proper : Proper ((≡) ==> iff) (@set_finite A C _).
  Proof. intros X Y HX; apply exist_proper. by setoid_rewrite HX. Qed.

  Lemma empty_finite : set_finite (∅ : C).
  Proof. by exists []; intros ?; rewrite elem_of_empty. Qed.
  Lemma singleton_finite (x : A) : set_finite ({[ x ]} : C).
  Proof. exists [x]; intros y ->%elem_of_singleton; left. Qed.
  Lemma union_finite X Y : set_finite X → set_finite Y → set_finite (X ∪ Y).
  Proof.
    intros [lX ?] [lY ?]; exists (lX ++ lY); intros x.
    rewrite elem_of_union, elem_of_app; naive_solver.
  Qed.
  Lemma union_finite_inv_l X Y : set_finite (X ∪ Y) → set_finite X.
  Proof. intros [l ?]; exists l; set_solver. Qed.
  Lemma union_finite_inv_r X Y : set_finite (X ∪ Y) → set_finite Y.
  Proof. intros [l ?]; exists l; set_solver. Qed.

  Global Instance set_infinite_subseteq :
    Proper ((⊆) ==> impl) (@set_infinite A C _).
  Proof. intros X Y HX ?. eapply pred_infinite_impl; set_solver. Qed.
  Global Instance set_infinite_proper : Proper ((≡) ==> iff) (@set_infinite A C _).
  Proof. intros X Y HX; apply forall_proper. by setoid_rewrite HX. Qed.

  Lemma union_infinite_l X Y : set_infinite X → set_infinite (X ∪ Y).
  Proof. intros Hinf xs. destruct (Hinf xs). set_solver. Qed.
  Lemma union_infinite_r X Y : set_infinite Y → set_infinite (X ∪ Y).
  Proof. intros Hinf xs. destruct (Hinf xs). set_solver. Qed.
End set_finite_infinite.

Section more_finite.
  Context `{Set_ A C}.
  Implicit Types X Y : C.

  Lemma intersection_finite_l X Y : set_finite X → set_finite (X ∩ Y).
  Proof. intros [l ?]; exists l; intros x [??]%elem_of_intersection; auto. Qed.
  Lemma intersection_finite_r X Y : set_finite Y → set_finite (X ∩ Y).
  Proof. intros [l ?]; exists l; intros x [??]%elem_of_intersection; auto. Qed.
  Lemma difference_finite X Y : set_finite X → set_finite (X ∖ Y).
  Proof. intros [l ?]; exists l; intros x [??]%elem_of_difference; auto. Qed.
  Lemma difference_finite_inv X Y `{∀ x, Decision (x ∈ Y)} :
    set_finite Y → set_finite (X ∖ Y) → set_finite X.
  Proof.
    intros [l ?] [k ?]; exists (l ++ k).
    intros x ?; destruct (decide (x ∈ Y)); rewrite elem_of_app; set_solver.
  Qed.

  Lemma difference_infinite X Y :
    set_infinite X → set_finite Y → set_infinite (X ∖ Y).
  Proof. intros Hinf [xs ?] xs'. destruct (Hinf (xs ++ xs')). set_solver. Qed.
End more_finite.

(** Sets of sequences of natural numbers *)
(* The set [seq_seq start len] of natural numbers contains the sequence
[start, start + 1, ..., start + (len-1)]. *)
Fixpoint set_seq `{Singleton nat C, Union C, Empty C} (start len : nat) : C :=
  match len with
  | O => ∅
  | S len' => {[ start ]} ∪ set_seq (S start) len'
  end.

Section set_seq.
  Context `{SemiSet nat C}.
  Implicit Types start len x : nat.

  Lemma elem_of_set_seq start len x :
    x ∈ set_seq (C:=C) start len ↔ start ≤ x < start + len.
  Proof.
    revert start. induction len as [|len IH]; intros start; simpl.
    - rewrite elem_of_empty. lia.
    - rewrite elem_of_union, elem_of_singleton, IH. lia.
  Qed.
  Global Instance set_unfold_seq start len x :
    SetUnfoldElemOf x (set_seq (C:=C) start len) (start ≤ x < start + len).
  Proof. constructor; apply elem_of_set_seq. Qed.

  Lemma set_seq_len_pos n start len : n ∈ set_seq (C:=C) start len → 0 < len.
  Proof. rewrite elem_of_set_seq. lia. Qed.

  Lemma set_seq_subseteq start1 len1 start2 len2 :
    0 < len1 →
    set_seq (C:=C) start1 len1 ⊆ set_seq (C:=C) start2 len2 ↔
      start2 ≤ start1 ∧ start1 + len1 ≤ start2 + len2.
  Proof.
    intros Hlen. set_unfold. split.
    - intros Hx. pose proof (Hx start1). pose proof (Hx (start1 + len1 - 1)). lia.
    - intros Heq x. lia.
  Qed.

  Lemma set_seq_subseteq_len_gt start1 len1 start2 len2 :
    set_seq (C:=C) start1 len1 ⊆ set_seq (C:=C) start2 len2 → len1 ≤ len2.
  Proof.
    destruct len1 as [|len1].
    - set_unfold. lia.
    - rewrite set_seq_subseteq; lia.
  Qed.

  Lemma set_seq_plus_disjoint start len1 len2 :
    set_seq (C:=C) start len1 ## set_seq (start + len1) len2.
  Proof. set_solver by lia. Qed.
  Lemma set_seq_plus start len1 len2 :
    set_seq (C:=C) start (len1 + len2)
    ≡ set_seq start len1 ∪ set_seq (start + len1) len2.
  Proof. set_solver by lia. Qed.
  Lemma set_seq_plus_L `{!LeibnizEquiv C} start len1 len2 :
    set_seq (C:=C) start (len1 + len2)
    = set_seq start len1 ∪ set_seq (start + len1) len2.
  Proof. unfold_leibniz. apply set_seq_plus. Qed.

  Lemma set_seq_S_start_disjoint start len :
    {[ start ]} ## set_seq (C:=C) (S start) len.
  Proof. set_solver by lia. Qed.
  Lemma set_seq_S_start start len :
    set_seq (C:=C) start (S len) ≡ {[ start ]} ∪ set_seq (S start) len.
  Proof. set_solver by lia. Qed.

  Lemma set_seq_S_end_disjoint start len :
    {[ start + len ]} ## set_seq (C:=C) start len.
  Proof. set_solver by lia. Qed.
  Lemma set_seq_S_end_union start len :
    set_seq start (S len) ≡@{C} {[ start + len ]} ∪ set_seq start len.
  Proof. set_solver by lia. Qed.
  Lemma set_seq_S_end_union_L `{!LeibnizEquiv C} start len :
    set_seq start (S len) =@{C} {[ start + len ]} ∪ set_seq start len.
  Proof. unfold_leibniz. apply set_seq_S_end_union. Qed.

  Lemma list_to_set_seq start len :
    list_to_set (seq start len) =@{C} set_seq start len.
  Proof. revert start; induction len; intros; f_equal/=; auto. Qed.

  Lemma set_seq_finite start len : set_finite (set_seq (C:=C) start len).
  Proof.
    exists (seq start len); intros x. rewrite <-list_to_set_seq. set_solver.
  Qed.
End set_seq.

(** Mimimal elements *)
Definition minimal `{ElemOf A C} (R : relation A) (x : A) (X : C) : Prop :=
  ∀ y, y ∈ X → R y x → R x y.
Instance: Params (@minimal) 5 := {}.
Typeclasses Opaque minimal.

Section minimal.
  Context `{SemiSet A C} {R : relation A}.
  Implicit Types X Y : C.

  Global Instance minimal_proper x : Proper ((≡@{C}) ==> iff) (minimal R x).
  Proof. intros X X' y; unfold minimal; set_solver. Qed.

  Lemma minimal_anti_symm_1 `{!AntiSymm (=) R} X x y :
    minimal R x X → y ∈ X → R y x → x = y.
  Proof. intros Hmin ??. apply (anti_symm _); auto. Qed.
  Lemma minimal_anti_symm `{!AntiSymm (=) R} X x :
    minimal R x X ↔ ∀ y, y ∈ X → R y x → x = y.
  Proof. unfold minimal; naive_solver eauto using minimal_anti_symm_1. Qed.

  Lemma minimal_strict_1 `{!StrictOrder R} X x y :
    minimal R x X → y ∈ X → ¬R y x.
  Proof. intros Hmin ??. destruct (irreflexivity R x); trans y; auto. Qed.
  Lemma minimal_strict `{!StrictOrder R} X x :
    minimal R x X ↔ ∀ y, y ∈ X → ¬R y x.
  Proof. unfold minimal; split; [eauto using minimal_strict_1|naive_solver]. Qed.

  Lemma empty_minimal x : minimal R x (∅ : C).
  Proof. unfold minimal; set_solver. Qed.
  Lemma singleton_minimal x : minimal R x ({[ x ]} : C).
  Proof. unfold minimal; set_solver. Qed.
  Lemma singleton_minimal_not_above y x : ¬R y x → minimal R x ({[ y ]} : C).
  Proof. unfold minimal; set_solver. Qed.
  Lemma union_minimal X Y x :
    minimal R x X → minimal R x Y → minimal R x (X ∪ Y).
  Proof. unfold minimal; set_solver. Qed.
  Lemma minimal_subseteq X Y x : minimal R x X → Y ⊆ X → minimal R x Y.
  Proof. unfold minimal; set_solver. Qed.

  Lemma minimal_weaken `{!Transitive R} X x x' :
    minimal R x X → R x' x → minimal R x' X.
  Proof.
    intros Hmin ? y ??. trans x; [done|]. by eapply (Hmin y), transitivity.
  Qed.
End minimal.


Require Coq.Logic.Classical_Prop.
Require Import ProofCheckingEuclid.euclidean_axioms.
Require Import ProofCheckingEuclid.euclidean_defs.
Require Import ProofCheckingEuclid.euclidean_tactics.
Require Import ProofCheckingEuclid.lemma_NCdistinct.
Require Import ProofCheckingEuclid.lemma_NCorder.
Require Import ProofCheckingEuclid.lemma_angleordertransitive.
Require Import ProofCheckingEuclid.lemma_betweennotequal.
Require Import ProofCheckingEuclid.lemma_collinear_ABC_ABD_BCD.
Require Import ProofCheckingEuclid.lemma_collinearorder.
Require Import ProofCheckingEuclid.lemma_congruenceflip.
Require Import ProofCheckingEuclid.lemma_inequalitysymmetric.
Require Import ProofCheckingEuclid.lemma_layoffunique.
Require Import ProofCheckingEuclid.lemma_onray_impliescollinear.
Require Import ProofCheckingEuclid.lemma_onray_shared_initial_point.
Require Import ProofCheckingEuclid.lemma_onray_strict.
Require Import ProofCheckingEuclid.lemma_orderofpoints_ABC_BCD_ACD.
Require Import ProofCheckingEuclid.lemma_s_n_col_ncol.
Require Import ProofCheckingEuclid.lemma_s_ncol_ABD_col_ABC_ncol_ACD.
Require Import ProofCheckingEuclid.lemma_s_ncol_n_col.
Require Import ProofCheckingEuclid.lemma_s_os.
Require Import ProofCheckingEuclid.lemma_s_ss.
Require Import ProofCheckingEuclid.lemma_sameside_onray.
Require Import ProofCheckingEuclid.lemma_samesidesymmetric.
Require Import ProofCheckingEuclid.proposition_07.

Section Euclid.

Context `{Ax:euclidean_neutral_ruler_compass}.

Lemma lemma_angletrichotomy :
	forall A B C D E F,
	LtA A B C D E F ->
	~ LtA D E F A B C.
Proof.
	intros A B C D E F.
	intros LtA_ABC_DEF.

	assert (~ LtA D E F A B C) as n_LtA_DEF_ABC.
	{
		intros LtA_DEF_ABC.

		pose proof (lemma_angleordertransitive _ _ _ _ _ _ _ _ _ LtA_ABC_DEF LtA_DEF_ABC) as LtA_ABC_ABC.

		destruct LtA_ABC_ABC as (G & H & J & BetS_G_H_J & OnRay_BA_G & OnRay_BC_J & CongA_ABC_ABH).

		pose proof (axiom_betweennesssymmetry _ _ _ BetS_G_H_J) as BetS_J_H_G.
		pose proof (lemma_betweennotequal _ _ _ BetS_G_H_J) as (_ & neq_G_H & _).
		pose proof (lemma_betweennotequal _ _ _ BetS_J_H_G) as (_ & neq_J_H & _).
		pose proof (lemma_inequalitysymmetric _ _ neq_G_H) as neq_H_G.
		pose proof (lemma_inequalitysymmetric _ _ neq_J_H) as neq_H_J.

		assert (Col G H J) as Col_G_H_J by (unfold Col; one_of_disjunct BetS_G_H_J).
		pose proof (lemma_collinearorder _ _ _ Col_G_H_J) as (_ & Col_H_J_G & _ & _ & _).

		pose proof (lemma_onray_strict _ _ _ OnRay_BA_G) as neq_B_G.
		pose proof (lemma_inequalitysymmetric _ _ neq_B_G) as neq_G_B.
		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BA_G) as Col_B_A_G.
		pose proof (lemma_collinearorder _ _ _ Col_B_A_G) as (Col_A_B_G & _ & Col_G_B_A  & Col_B_G_A  & _).

		pose proof (lemma_onray_strict _ _ _ OnRay_BC_J) as neq_B_J.
		pose proof (lemma_inequalitysymmetric _ _ neq_B_J) as neq_J_B.
		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BC_J) as Col_B_C_J.
		pose proof (lemma_collinearorder _ _ _ Col_B_C_J) as (_ & _ & Col_J_B_C & Col_B_J_C  & _).

		destruct CongA_ABC_ABH as (U & V & u & v & OnRay_BA_U & OnRay_BC_V & OnRay_BA_u & OnRay_BH_v & Cong_BU_Bu & Cong_BV_Bv & Cong_UV_uv & nCol_A_B_C).

		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BA_U) as Col_B_A_U.
		pose proof (lemma_collinearorder _ _ _ Col_B_A_U) as (_ & _ & Col_U_B_A & _ & _).
		pose proof (lemma_onray_strict _ _ _ OnRay_BA_U) as neq_B_U.
		pose proof (lemma_inequalitysymmetric _ _ neq_B_U) as neq_U_B.
		pose proof (lemma_onray_shared_initial_point _ _ _ _ OnRay_BA_G OnRay_BA_U) as OnRay_BG_U.
		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BG_U) as Col_B_G_U.
		pose proof (lemma_collinearorder _ _ _ Col_B_G_U) as (_ & _ & _ & Col_B_U_G & _).

		pose proof (lemma_layoffunique _ _ _ _ OnRay_BA_U OnRay_BA_u Cong_BU_Bu) as eq_U_u.
		assert (Cong U V U v) as Cong_UV_Uv by (rewrite eq_U_u at 2; exact Cong_UV_uv).
		pose proof (lemma_congruenceflip _ _ _ _ Cong_UV_Uv) as (Cong_VU_vU & _ & _).

		pose proof (lemma_onray_shared_initial_point _ _ _ _ OnRay_BC_J OnRay_BC_V) as OnRay_BJ_V.
		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BJ_V) as Col_B_J_V.
		pose proof (lemma_collinearorder _ _ _ Col_B_J_V) as (_ & _ & Col_V_B_J & _ & _).

		pose proof (lemma_congruenceflip _ _ _ _ Cong_BV_Bv) as (Cong_VB_vB & _ & _).

		pose proof (lemma_s_ncol_n_col _ _ _ nCol_A_B_C) as n_Col_A_B_C.
		pose proof (lemma_NCdistinct _ _ _ nCol_A_B_C) as (neq_A_B & _ & _ & _ & _ & _).
		pose proof (lemma_NCorder _ _ _ nCol_A_B_C) as (nCol_B_A_C & nCol_B_C_A & _ & _ & _).

		pose proof (lemma_s_ncol_ABD_col_ABC_ncol_ACD _ _ _ _ nCol_B_C_A Col_B_C_J neq_B_J) as nCol_B_J_A.
		pose proof (lemma_NCorder _ _ _ nCol_B_J_A) as (_ & _ & _ & nCol_B_A_J & _).
		pose proof (lemma_s_ncol_n_col _ _ _ nCol_B_A_J) as n_Col_B_A_J.

		pose proof (postulate_Euclid2 _ _ neq_H_G) as (P & BetS_H_G_P).

		pose proof (lemma_orderofpoints_ABC_BCD_ACD _ _ _ _ BetS_J_H_G BetS_H_G_P) as BetS_J_G_P.

		pose proof (lemma_s_os _ _ _ _ _ BetS_J_G_P Col_B_A_G nCol_B_A_J) as OS_J_BA_P.

		assert (~ Col B U H) as n_Col_B_U_H.
		{
			intros Col_B_U_H.

			pose proof (lemma_collinearorder _ _ _ Col_B_U_H) as (Col_U_B_H & _ & _ & _ & _).

			pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_U_B_A Col_U_B_H neq_U_B) as Col_B_A_H.
			pose proof (lemma_collinearorder _ _ _ Col_B_A_H) as (Col_A_B_H & _ & Col_H_B_A & _ & _).

			pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_A_B_G Col_A_B_H neq_A_B) as Col_B_G_H.
			pose proof (lemma_collinearorder _ _ _ Col_B_G_H) as (_ & Col_G_H_B & _ & _ & _).

			pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_G_H_B Col_G_H_J neq_G_H) as Col_H_B_J.

			assert (~ neq H B) as eq_H_B.
			{
				intros neq_H_B.

				pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_H_B_J Col_H_B_A neq_H_B) as Col_B_J_A.
				pose proof (lemma_collinearorder _ _ _ Col_B_J_A) as (_ & _ & _ & Col_B_A_J & _).

				contradict Col_B_A_J.
				exact n_Col_B_A_J.
			}
			apply Classical_Prop.NNPP in eq_H_B.

			assert (BetS G B J) as BetS_G_B_J by (rewrite <- eq_H_B; exact BetS_G_H_J).
			pose proof (lemma_betweennotequal _ _ _ BetS_G_B_J) as (_ & _ & neq_G_J).

			assert (Col G B J) as Col_G_B_J by (unfold Col; one_of_disjunct BetS_G_B_J).
			pose proof (lemma_collinearorder _ _ _ Col_G_B_J) as (Col_B_G_J & _ & _ & Col_G_J_B & _).

			pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_B_G_J Col_B_G_A neq_B_G) as Col_G_J_A.
			pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_G_J_A Col_G_J_B neq_G_J) as Col_J_A_B.
			pose proof (lemma_collinearorder _ _ _ Col_J_A_B) as (_ & _ & _ & _ & Col_B_A_J).

			contradict Col_B_A_J.
			exact n_Col_B_A_J.
		}
		pose proof (lemma_s_n_col_ncol _ _ _ n_Col_B_U_H) as nCol_B_U_H.

		pose proof (lemma_s_os _ _ _ _ _ BetS_H_G_P Col_B_U_G nCol_B_U_H) as OS_H_BU_P.

		pose proof (lemma_s_ncol_ABD_col_ABC_ncol_ACD _ _ _ _ nCol_B_A_C Col_B_A_U neq_B_U) as nCol_B_U_C.
		pose proof (lemma_NCorder _ _ _ nCol_B_U_C) as (_ & _ & _ & nCol_B_C_U & _).
		pose proof (lemma_s_ncol_ABD_col_ABC_ncol_ACD _ _ _ _ nCol_B_C_U Col_B_C_J neq_B_J) as nCol_B_J_U.
		pose proof (lemma_NCorder _ _ _ nCol_B_J_U) as (_ & _ & _ & nCol_B_U_J & _).

		pose proof (lemma_s_ss _ _ _ _ _ _ _ Col_B_U_G Col_B_U_G BetS_J_G_P BetS_H_G_P nCol_B_U_J nCol_B_U_H) as SS_J_H_BU.
		pose proof (lemma_samesidesymmetric _ _ _ _ SS_J_H_BU) as (SS_H_J_BU & _ & _).
		assert (eq B B) as eq_B_B by (reflexivity).
		assert (Col B B U) as Col_B_B_U by (unfold Col; one_of_disjunct eq_B_B).
		pose proof (lemma_sameside_onray _ _ _ _ _ _ SS_H_J_BU Col_B_B_U OnRay_BJ_V) as SS_H_V_BU.
		pose proof (lemma_samesidesymmetric _ _ _ _ SS_H_V_BU) as (SS_V_H_BU & _ & _).
		pose proof (lemma_sameside_onray _ _ _ _ _ _ SS_V_H_BU Col_B_B_U OnRay_BH_v) as SS_V_v_BU.

		pose proof (proposition_07 _ _ _ _ neq_B_U Cong_VB_vB Cong_VU_vU SS_V_v_BU) as eq_V_v.
		assert (OnRay B H V) as OnRay_BH_V by (rewrite eq_V_v; exact OnRay_BH_v).
		pose proof (lemma_onray_strict _ _ _ OnRay_BH_V) as neq_B_V.
		pose proof (lemma_inequalitysymmetric _ _ neq_B_V) as neq_V_B.
		pose proof (lemma_onray_impliescollinear _ _ _ OnRay_BH_V) as Col_B_H_V.
		pose proof (lemma_collinearorder _ _ _ Col_B_H_V) as (_ & _ & Col_V_B_H & _ & _).

		pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_V_B_J Col_V_B_H neq_V_B) as Col_B_J_H.
		pose proof (lemma_collinearorder _ _ _ Col_B_J_H) as (_ & _ & _ & _ & Col_H_J_B).
		pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_H_J_B Col_H_J_G neq_H_J) as Col_J_B_G.
		pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_J_B_G Col_J_B_C neq_J_B) as Col_B_G_C.
		pose proof (lemma_collinearorder _ _ _ Col_B_G_C) as (Col_G_B_C & _ & _ & _ & _).
		pose proof (lemma_collinear_ABC_ABD_BCD _ _ _ _ Col_G_B_C Col_G_B_A neq_G_B) as Col_B_C_A.
		pose proof (lemma_collinearorder _ _ _ Col_B_C_A) as (_ & _ & Col_A_B_C & _ & _).

		contradict Col_A_B_C.
		exact n_Col_A_B_C.
	}
	exact n_LtA_DEF_ABC.
Qed.

End Euclid.


(* This program is free software; you can redistribute it and/or      *)
(* modify it under the terms of the GNU Lesser General Public License *)
(* as published by the Free Software Foundation; either version 2.1   *)
(* of the License, or (at your option) any later version.             *)
(*                                                                    *)
(* This program is distributed in the hope that it will be useful,    *)
(* but WITHOUT ANY WARRANTY; without even the implied warranty of     *)
(* MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the      *)
(* GNU General Public License for more details.                       *)
(*                                                                    *)
(* You should have received a copy of the GNU Lesser General Public   *)
(* License along with this program; if not, write to the Free         *)
(* Software Foundation, Inc., 51 Franklin St, Fifth Floor, Boston, MA *)
(* 02110-1301 USA                                                     *)

Require Export ZArithRing.
Require Export ZArith.

Module Type Carrier.
Parameter A : Set.
Parameter Aopp : A -> A.
Parameters (Aplus Aminus Amult: A -> A -> A).
Parameters (A0 : A) (A1 : A).

Axiom A_ring : ring_theory A0 A1 Aplus Amult Aminus Aopp (eq(A:=A)).

End Carrier.

Module Zc : Carrier.

Definition A := Z.
Definition Aopp := Zopp.
Definition Aplus := Zplus.
Definition Aminus := Zminus.
Definition Amult := Zmult.
Definition A0 := 0%Z.
Definition A1 := 1%Z.

Definition A_ring := InitialRing.Zth.

End Zc.

  (*
  Local Variables:
  mode: coq
  coq-prog-name: "/usr/local/coq-7.4/bin/coqtop -emacs -q"
  compile-command: "make carrier.vo"
  End:
  *)


Require Import Tactics.
Require Import Setoid.

Require Import Logic.ClassicalFacts.

Axiom excluded_middle: excluded_middle.
Lemma rewrite_not_conj P Q: not (P /\ Q) <-> (not P) \/ (not Q).
Proof.
  split; intros; auto.
  - destruct (excluded_middle P).
    + right. intro. auto.
    + left. auto.
  - intro. destruct H0. destruct H; auto.
Qed.
Lemma rewrite_not_not P: not (not P) <-> P.
Proof.
  split; intros; auto.
  - destruct (excluded_middle P); auto.
    exfalso. auto.
Qed.

Definition key := nat.
Inductive treei (V : Type) : Type :=
| E
| T (k : key) (l : treei V) (v : V) (r : treei V).
Arguments E {V}.
Arguments T {V}.

Inductive treei_head : treei nat -> nat -> Prop :=
| thd1: forall u u' l k r, u = u' -> treei_head (T l k u r) u'.

Inductive treei_member: treei nat -> nat -> Prop :=
| tmem1: forall u u' l k r, u = u' -> treei_member (T l k u r) u'
| tmem2: forall u v l k r, treei_member l v -> treei_member (T k l u r) v
| tmem3: forall u v l k r, treei_member r v -> treei_member (T k l u r) v.

Inductive treei_left: treei nat -> nat -> nat -> Prop :=
| tl1: forall u u' v v' l k r,
    u = u' -> v = v' -> treei_member l v -> treei_left (T k l u r) u' v'
| tl2: forall u v w l k r, treei_left l u v -> treei_left (T k l w r) u v
| tl3: forall u v w l k r, treei_left r u v -> treei_left (T k l w r) u v.

Inductive treei_right: treei nat -> nat -> nat -> Prop :=
| tr1: forall u u' v v' l k r,
    u = u' -> v = v' -> treei_member r v -> treei_right (T k l u r) u' v'
| tr2: forall u v w l k r, treei_right l u v -> treei_right (T k l w r) u v
| tr3: forall u v w l k r, treei_right r u v -> treei_right (T k l w r) u v.

Definition treei_ancestor (t: treei nat) (u: nat) (v: nat): Prop :=
  (treei_left t u v) \/ (treei_right t u v).

Inductive treei_parallel: treei nat -> nat -> nat -> Prop :=
| tp1: forall u u' v v' w l k r,
    u = u' -> v = v' -> treei_member l u -> treei_member r v ->
    treei_parallel (T k l w r) u' v'
| tp2: forall u v w l k r, treei_parallel l u v -> treei_parallel (T k l w r) u v
| tp3: forall u v w l k r, treei_parallel r u v -> treei_parallel (T k l w r) u v.

Hint Constructors treei_head: core.
Hint Constructors treei_member: core.
Hint Constructors treei_left: core.
Hint Constructors treei_right: core.
Hint Constructors treei_parallel: core.

Hint Unfold not: core.
Hint Unfold treei_ancestor: core.

Lemma thd_lmem1: forall (t: treei nat) (u: nat), treei_head t u -> treei_member t u.
Proof. intros. induction H; subst; auto. Qed.

Hint Resolve thd_lmem1: core.

Lemma thd: forall (l: treei nat) (u: nat) (v: nat), treei_head l u -> treei_head l v -> u = v.
Proof.
  intros. destruct l.
  - inversion H.
  - inversion H; inversion H0; subst; auto.
Qed.

Hint Resolve thd: core.

Lemma tl_lmem1: forall (l: treei nat) (u: nat) (v: nat), treei_left l u v -> treei_member l u.
Proof. intros. induction H; subst; auto. Qed.

Lemma tl_lmem2: forall (l: treei nat) (u: nat) (v: nat), treei_left l u v -> treei_member l v.
Proof. intros. induction H; subst; auto. Qed.

Lemma tr_lmem1: forall (l: treei nat) (u: nat) (v: nat), treei_right l u v -> treei_member l u.
Proof. intros. induction H; subst; auto. Qed.

Lemma tr_lmem2: forall (l: treei nat) (u: nat) (v: nat), treei_right l u v -> treei_member l v.
Proof. intros. induction H; subst; auto. Qed.

Lemma tp_lmem1: forall (l: treei nat) (u: nat) (v: nat), treei_parallel l u v -> treei_member l u.
Proof. intros. induction H; subst; auto. Qed.

Lemma tp_lmem2: forall (l: treei nat) (u: nat) (v: nat), treei_parallel l u v -> treei_member l v.
Proof. intros. induction H; subst; auto. Qed.

Hint Resolve tl_lmem1: core.
Hint Resolve tl_lmem2: core.
Hint Resolve tr_lmem1: core.
Hint Resolve tr_lmem2: core.
Hint Resolve tp_lmem1: core.
Hint Resolve tp_lmem2: core.

Lemma rewrite_not_disj P Q: not (P \/ Q) <-> (not P) /\ (not Q).
Proof. split; auto. intros. intro. destruct H. destruct H0; contradiction. Qed.

Ltac invsub H := inversion H; subst.

Definition it (b:nat) (x:nat) (l:treei nat) (r: treei nat) :=
  T b l x r.

Inductive it_spec: nat -> nat -> treei nat -> treei nat -> treei nat -> Prop :=
| canonical_t_spec: forall b l x r t,
    (forall u, treei_head t u <-> u = x) ->
    (forall u, treei_member t u <-> ((u = x) \/ (treei_member l u) \/ (treei_member r u))) ->
    (forall u v, treei_left t u v <->
            ((treei_left l u v) \/ (treei_left r u v) \/ ((u = x) /\ (treei_member l v)))) ->
    (forall u v, treei_right t u v <->
                 ((treei_right l u v) \/ (treei_right r u v) \/ ((u = x) /\ (treei_member r v)))) ->
    (forall u v, treei_parallel t u v <->
            ((treei_parallel l u v) \/ (treei_parallel r u v) \/ ((treei_member l u) /\ (treei_member r v)))) ->
    it_spec b x l r t.

Lemma it_it_spec: forall (k: nat) (l: treei nat) (x: nat) (r: treei nat) (t: treei nat),
    it k x l r = t -> it_spec k x l r t.
Proof.
  unfold it.
  intros; subst; apply canonical_t_spec; intro.
  - split; auto. intro. invsub H; auto.
  - split; intros; auto; invsub H; auto. destruct_pairs. subst.
    destruct H; auto. destruct H; auto.
  - split; intros; auto. invsub H; auto. destruct H; auto. destruct H; auto. destruct H; eauto.
  - split; intros; auto. invsub H; auto. destruct H; auto. destruct H; auto. destruct H; eauto.
  - split; intros; auto. invsub H; auto. destruct H. apply tp2; auto. destruct H. apply tp3; auto. destruct H; apply tp1 with u v; auto.
Qed.

Definition rank (tr: treei nat): nat :=
  match tr with
  | E => 0
  | T r _ _ _ => r
  end.

Definition imaket (x:nat) (a:treei nat) (b: treei nat) :=
  if Nat.leb (rank b) (rank a) then T (rank b + 1) a x b
  else T (rank a + 1) a x b.

Definition imaket_t (x:nat) (a:treei nat) (b: treei nat) : exists k,
    imaket x a b = it k x a b.
Proof.
  unfold it.
  unfold imaket. unfold rank.
  induction a, b; simpl.
  - exists 1. auto.
  - destruct k; simpl.
    + exists 1; auto.
    + exists 1; auto.
  - exists 1. auto.
  - destruct (PeanoNat.Nat.leb_spec0 k0 k).
    + exists (k0 + 1). auto.
    + exists (k + 1). auto.
Qed.

Inductive imaket_spec: nat -> treei nat -> treei nat -> treei nat -> Prop :=
| canonical_maket_spec: forall l x r t,
    (forall u, treei_head t u <-> u = x) ->
    (forall u, treei_member t u <-> ((u = x) \/ (treei_member l u) \/ (treei_member r u))) ->
    (forall u v, treei_left t u v <->
            ((treei_left l u v) \/ (treei_left r u v) \/ ((u = x) /\ (treei_member l v)))) ->
    (forall u v, treei_right t u v <->
                 ((treei_right l u v) \/ (treei_right r u v) \/ ((u = x) /\ (treei_member r v)))) ->
    (forall u v, treei_parallel t u v <->
            ((treei_parallel l u v) \/ (treei_parallel r u v) \/ ((treei_member l u) /\ (treei_member r v)))) ->
    imaket_spec x l r t.

Ltac not_empty :=
  match goal with
  | H: treei_head E _ \/ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_member E _ \/ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_left E _ _ \/ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_right E _ _ \/ _|- _ => destruct H; subst; eauto; try not_empty
  | H: treei_parallel E _ _ \/ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ treei_head E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ treei_member E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ treei_left E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ treei_right E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ treei_parallel E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ _ \/ treei_head E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ _ \/ treei_member E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ _ \/ treei_left E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ _ \/ treei_right E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ \/ _ \/ treei_parallel E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_head E _ /\ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_member E _ /\ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_left E _ _ /\ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_right E _ _ /\ _|- _ => destruct H; subst; eauto; try not_empty
  | H: treei_parallel E _ _ /\ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ /\ treei_head E _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ /\ treei_member E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ /\ treei_left E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ /\ treei_right E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: _ /\ treei_parallel E _ _ |- _ => destruct H; subst; eauto; try not_empty
  | H: treei_head E _ |- _ => inversion H
  | H: treei_member E _ |- _ => inversion H
  | H: treei_left E _ _ |- _ => inversion H
  | H: treei_right E _ _ |- _ => inversion H
  | H: treei_parallel E _ _ |- _ => inversion H
  end.

Lemma imaket_imaket_spec: forall (l: treei nat) (x: nat) (r: treei nat) (t: treei nat),
    imaket x l r = t -> imaket_spec x l r t.
Proof.
  intros. destruct (imaket_t x l r).
  rewrite H0 in H. clear H0. apply it_it_spec in H. inversion H; subst.
  apply canonical_maket_spec; auto.
Qed.

Definition ie : treei nat := E.

Inductive ie_spec: treei nat -> Prop :=
| canonical_e_spec: forall t,
    (forall u, not (treei_head t u)) ->
    (forall u, not (treei_member t u)) ->
    (forall u v, not (treei_left t u v)) ->
    (forall u v, not (treei_right t u v)) ->
    ie_spec t.

Lemma ie_ie_spec: ie_spec ie.
Proof.
  intros; subst; apply canonical_e_spec; intros; try (intro; inversion H).
Qed.

Ltac destruct_disj :=
 match goal with
 | [H : (_ \/ _) |- _] => destruct H
 | [H : _ |- (_ /\ _)] => split
 end.

Ltac solve_eq :=
  match goal with
  | [H : _ |- (_ = _) ] => (try symmetry; eauto); eauto
  end.

Ltac solve_not :=
  match goal with
  | [H : _ |- (not _) ] => intro Z
  end; subst; eauto.

Ltac solve_disjs := repeat (destruct_disj);
                    destruct_pairs; subst;
                    try contradiction; try solve_eq; try solve_not; eauto;
                    try (exfalso; eauto).

Ltac solve_it :=
  intro btspec;
  inversion btspec as [a b c d e Hhd Hmem Hl Hr Hp]; subst;
  try setoid_rewrite Hhd; try setoid_rewrite Hmem; try setoid_rewrite Hl; try setoid_rewrite Hr; try setoid_rewrite Hp;
  clear Hhd Hmem Hl Hr Hp;
  repeat ((repeat setoid_rewrite rewrite_not_conj);
          (repeat setoid_rewrite rewrite_not_disj);
          (repeat setoid_rewrite rewrite_not_not));
  intros; destruct_pairs; eauto; try solve_disjs.

Ltac solve_imaket :=
  intro btspec;
  inversion btspec as [b c d e Hhd Hmem Hl Hr Hp];  try unfold treei_ancestor; subst;
  try setoid_rewrite Hhd; try setoid_rewrite Hmem; try setoid_rewrite Hl; try setoid_rewrite Hr; try setoid_rewrite Hp;
  clear Hhd Hmem Hl Hr Hp;
  repeat ((repeat setoid_rewrite rewrite_not_conj);
          (repeat setoid_rewrite rewrite_not_disj);
          (repeat setoid_rewrite rewrite_not_not));
  intros; destruct_pairs; eauto; try solve_disjs.

Ltac solve_ie :=
  intro espec;
  inversion espec as [a Hhd Hmem Hl Hr]; subst;
  try setoid_rewrite Hhd; try setoid_rewrite Hmem; try setoid_rewrite Hl; try setoid_rewrite Hr; intros;
    try apply Hhd;
    try apply Hmem;
    try apply Hl;
    try apply Hr;
    eauto.


(*************************************************************************

 Eilenberg-Moore objects of comonads

 In this file, we define the notion of Eilenberg-Moore object for
 comonads.
 The main purpose of this notion, is that Eilenberg-Moore objects for
 comonads in `B` give rise to Eilenberg-Moore objects in `op2_bicat B`.

 Contents
 1. Eilenberg-Moore objects of comonads via universal mapping properties
 2. Being an Eilenberg-Moore object for a comonad is a proposition
 3. Bicategories with Eilenberg-Moore objects for comonads

 *************************************************************************)
Require Import UniMath.Foundations.All.
Require Import UniMath.MoreFoundations.All.
Require Import UniMath.CategoryTheory.Core.Categories.
Require Import UniMath.Bicategories.Core.Bicat.
Import Bicat.Notations.
Require Import UniMath.Bicategories.Core.Invertible_2cells.
Require Import UniMath.Bicategories.Core.BicategoryLaws.
Require Import UniMath.Bicategories.Core.Unitors.
Require Import UniMath.Bicategories.Core.TransportLaws.
Require Import UniMath.Bicategories.Core.Univalence.
Require Import UniMath.Bicategories.Core.Examples.OpCellBicat.
Require Import UniMath.Bicategories.DisplayedBicats.DispBicat.
Import DispBicat.Notations.
Require Import UniMath.Bicategories.DisplayedBicats.Examples.MonadsLax.

Local Open Scope cat.

Section EilenbergMooreComonad.
  Context {B : bicat}
          (m : mnd (op2_bicat B)).

  Let z : B := ob_of_mnd m.
  Let h : z --> z := endo_of_mnd m.
  Let ε : h ==> id₁ _ := unit_of_mnd m.
  Let ν : h ==> h · h := mult_of_mnd m.

  (**
   1. Eilenberg-Moore objects of comonads via universal mapping properties
   *)
  Definition em_comnd_cone
    : UU
    := ∑ (x : B)
         (f : x --> z)
         (γ : f ==> f · h),
      (γ • (f ◃ ε) = rinvunitor _)
      ×
      (γ • (γ ▹ _) • rassociator _ _ _ = γ • (f ◃ ν)).

  Definition make_em_comnd_cone
             {x : B}
             (f : x --> z)
             (γ : f ==> f · h)
             (fε : γ • (f ◃ ε) = rinvunitor _)
             (fν : γ • (γ ▹ _) • rassociator _ _ _ = γ • (f ◃ ν))
    : em_comnd_cone
    := x ,, f ,, γ ,, fε ,, fν.

  Coercion em_comnd_cone_ob (q : em_comnd_cone) : B := pr1 q.

  Section Projections.
    Context (q : em_comnd_cone).

    Definition mor_of_em_comnd_cone
      : q --> z
      := pr12 q.

    Definition cell_of_em_comnd_cone
      : mor_of_em_comnd_cone
        ==>
        mor_of_em_comnd_cone · h
      := pr122 q.

    Definition em_comnd_cone_counit
      : cell_of_em_comnd_cone • (_ ◃ ε) = rinvunitor _
      := pr1 (pr222 q).

    Definition em_comnd_cone_comult
      : (cell_of_em_comnd_cone
         • (cell_of_em_comnd_cone ▹ _))
         • rassociator _ _ _
        =
        cell_of_em_comnd_cone
        • (_ ◃ ν)
      := pr2 (pr222 q).
  End Projections.

  Definition em_comnd_cone_mor
             (q₁ q₂ : em_comnd_cone)
    : UU
    := ∑ (f : q₁ --> q₂)
         (α : mor_of_em_comnd_cone q₁ ==> f · mor_of_em_comnd_cone q₂),
       (cell_of_em_comnd_cone q₁ • (α ▹ _)
        =
        α • (_ ◃ cell_of_em_comnd_cone q₂) • lassociator _ _ _)
       ×
       is_invertible_2cell α.

  Definition make_em_comnd_cone_mor
             {q₁ q₂ : em_comnd_cone}
             (f : q₁ --> q₂)
             (α : mor_of_em_comnd_cone q₁ ==> f · mor_of_em_comnd_cone q₂)
             (p : cell_of_em_comnd_cone q₁ • (α ▹ _)
                  =
                  α • (_ ◃ cell_of_em_comnd_cone q₂) • lassociator _ _ _)
             (Hα : is_invertible_2cell α)
    : em_comnd_cone_mor q₁ q₂
    := f ,, α ,, p ,, Hα.

  Coercion mor_of_em_comnd_cone_mor
           {q₁ q₂ : em_comnd_cone}
           (f : em_comnd_cone_mor q₁ q₂)
    : q₁ --> q₂
    := pr1 f.

  Definition cell_of_em_comnd_cone_mor
             {q₁ q₂ : em_comnd_cone}
             (f : em_comnd_cone_mor q₁ q₂)
    : mor_of_em_comnd_cone q₁ ==> f · mor_of_em_comnd_cone q₂
    := pr12 f.

  Definition em_comnd_cone_mor_endo
             {q₁ q₂ : em_comnd_cone}
             (f : em_comnd_cone_mor q₁ q₂)
    : cell_of_em_comnd_cone q₁
      • (cell_of_em_comnd_cone_mor f ▹ _)
      =
      cell_of_em_comnd_cone_mor f
      • (_ ◃ cell_of_em_comnd_cone q₂)
      • lassociator _ _ _
    := pr122 f.

  Definition cell_of_em_comnd_cone_mor_is_invertible
             {q₁ q₂ : em_comnd_cone}
             (f : em_comnd_cone_mor q₁ q₂)
    : is_invertible_2cell (cell_of_em_comnd_cone_mor f)
    := pr222 f.

  Definition inv2cell_of_em_comnd_cone_mor
             {q₁ q₂ : em_comnd_cone}
             (f : em_comnd_cone_mor q₁ q₂)
    : invertible_2cell
        (mor_of_em_comnd_cone q₁)
        (f · mor_of_em_comnd_cone q₂).
  Proof.
    use make_invertible_2cell.
    - exact (cell_of_em_comnd_cone_mor f).
    - exact (cell_of_em_comnd_cone_mor_is_invertible f).
  Defined.

  Definition path_em_comnd_cone_mor
             (HB : is_univalent_2_1 B)
             {q₁ q₂ : em_comnd_cone}
             {f₁ f₂ : em_comnd_cone_mor q₁ q₂}
             (α : invertible_2cell f₁ f₂)
             (p : cell_of_em_comnd_cone_mor f₁ • (α ▹ mor_of_em_comnd_cone q₂)
                  =
                  cell_of_em_comnd_cone_mor f₂)
    : f₁ = f₂.
  Proof.
    use total2_paths_f.
    - exact (isotoid_2_1 HB α).
    - use subtypePath.
      {
        intro.
        apply isapropdirprod ; [ apply cellset_property | apply isaprop_is_invertible_2cell ].
      }
      rewrite pr1_transportf.
      rewrite transport_two_cell_FlFr.
      rewrite maponpaths_for_constant_function.
      cbn.
      rewrite id2_left.
      rewrite <- idtoiso_2_1_rwhisker.
      rewrite idtoiso_2_1_isotoid_2_1.
      exact p.
  Qed.

  Definition has_em_comnd_ump_1
             (e : em_comnd_cone)
    : UU
    := ∏ (q : em_comnd_cone), em_comnd_cone_mor q e.

  Definition has_em_comnd_ump_2
             (e : em_comnd_cone)
    : UU
    := ∏ (x : B)
         (g₁ g₂ : x --> e)
         (α : g₁ · mor_of_em_comnd_cone e ==> g₂ · mor_of_em_comnd_cone e)
         (p : α
              • (_ ◃ cell_of_em_comnd_cone e)
              • lassociator _ _ _
              =
              (_ ◃ cell_of_em_comnd_cone e)
              • lassociator _ _ _
              • (α ▹ _)),
       ∃! (β : g₁ ==> g₂), β ▹ _ = α.

  Definition has_em_comnd_ump
             (e : em_comnd_cone)
    : UU
    := has_em_comnd_ump_1 e × has_em_comnd_ump_2 e.

  Section Projections.
    Context {e : em_comnd_cone}
            (He : has_em_comnd_ump e).

    Definition em_comnd_ump_mor
               {a : B}
               (g : a --> z)
               (γ : g ==> g · h)
               (p₁ : γ • (g ◃ ε) = rinvunitor g)
               (p₂ : (γ • (γ ▹ h)) • rassociator g h h = γ • (g ◃ ν))
      : a --> e
      := pr1 He (make_em_comnd_cone g γ p₁ p₂).

    Definition em_comnd_ump_mor_cell
               {a : B}
               (g : a --> z)
               (γ : g ==> g · h)
               (p₁ : γ • (g ◃ ε) = rinvunitor g)
               (p₂ : (γ • (γ ▹ h)) • rassociator g h h = γ • (g ◃ ν))
      : g ==> em_comnd_ump_mor g γ p₁ p₂ · mor_of_em_comnd_cone e
      := cell_of_em_comnd_cone_mor (pr1 He (make_em_comnd_cone g γ p₁ p₂)).

    Definition em_comnd_ump_mor_cell_endo
               {a : B}
               (g : a --> z)
               (γ : g ==> g · h)
               (p₁ : γ • (g ◃ ε) = rinvunitor g)
               (p₂ : (γ • (γ ▹ h)) • rassociator g h h = γ • (g ◃ ν))
      : γ
        • (em_comnd_ump_mor_cell g γ p₁ p₂ ▹ h)
        =
        em_comnd_ump_mor_cell g γ p₁ p₂
        • (em_comnd_ump_mor g γ p₁ p₂ ◃ cell_of_em_comnd_cone e)
        • lassociator _ _ _
      := em_comnd_cone_mor_endo (pr1 He (make_em_comnd_cone g γ p₁ p₂)).

    Definition em_comnd_ump_mor_cell_is_invertible
               {a : B}
               (g : a --> z)
               (γ : g ==> g · h)
               (p₁ : γ • (g ◃ ε) = rinvunitor g)
               (p₂ : (γ • (γ ▹ h)) • rassociator g h h = γ • (g ◃ ν))
      : is_invertible_2cell (em_comnd_ump_mor_cell g γ p₁ p₂)
      := cell_of_em_comnd_cone_mor_is_invertible
           (pr1 He (make_em_comnd_cone g γ p₁ p₂)).

    Definition em_comnd_ump_mor_inv2cell
               {a : B}
               (g : a --> z)
               (γ : g ==> g · h)
               (p₁ : γ • (g ◃ ε) = rinvunitor g)
               (p₂ : (γ • (γ ▹ h)) • rassociator g h h = γ • (g ◃ ν))
      : invertible_2cell
          g
          (em_comnd_ump_mor g γ p₁ p₂ · mor_of_em_comnd_cone e).
    Proof.
      use make_invertible_2cell.
      - exact (em_comnd_ump_mor_cell g γ p₁ p₂).
      - exact (em_comnd_ump_mor_cell_is_invertible g γ p₁ p₂).
    Defined.

    Definition em_comnd_ump_cell
               {x : B}
               {g₁ g₂ : x --> e}
               (α : g₁ · mor_of_em_comnd_cone e ==> g₂ · mor_of_em_comnd_cone e)
               (p : α
                    • (g₂ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    =
                    (g₁ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    • (α ▹ h))
      : g₁ ==> g₂
      := pr11 (pr2 He x g₁ g₂ α p).

    Definition em_comnd_ump_cell_eq
               {x : B}
               {g₁ g₂ : x --> e}
               (α : g₁ · mor_of_em_comnd_cone e ==> g₂ · mor_of_em_comnd_cone e)
               (p : α
                    • (g₂ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    =
                    (g₁ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    • (α ▹ h))
      : em_comnd_ump_cell α p ▹ _ = α
      := pr21 (pr2 He x g₁ g₂ α p).

    Definition em_comnd_ump_eq
               {x : B}
               {g₁ g₂ : x --> e}
               (α : g₁ · mor_of_em_comnd_cone e ==> g₂ · mor_of_em_comnd_cone e)
               (p : α
                    • (g₂ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    =
                    (g₁ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                    • (α ▹ h))
               {β₁ β₂ : g₁ ==> g₂}
               (q₁ : β₁ ▹ _ = α)
               (q₂ : β₂ ▹ _ = α)
      : β₁ = β₂.
    Proof.
      exact (maponpaths
               pr1
               (proofirrelevance
                  _
                  (isapropifcontr (pr2 He x g₁ g₂ α p))
                  (β₁ ,, q₁)
                  (β₂ ,, q₂))).
    Qed.

    Section Invertible.
      Context {x : B}
              {g₁ g₂ : x --> e}
              (α : invertible_2cell
                     (g₁ · mor_of_em_comnd_cone e)
                     (g₂ · mor_of_em_comnd_cone e))
              (p : α
                   • (g₂ ◃ cell_of_em_comnd_cone e)
                   • lassociator _ _ _
                   =
                   (g₁ ◃ cell_of_em_comnd_cone e)
                   • lassociator _ _ _
                   • (α ▹ h)).

      Definition em_comnd_ump_cell_inv
        : g₂ ==> g₁.
      Proof.
        refine (em_comnd_ump_cell (α^-1) _).
        abstract
          (use vcomp_move_L_Mp ; [ is_iso | ] ;
           rewrite !vassocl ;
           use vcomp_move_R_pM ; [ is_iso | ] ;
           cbn ;
           rewrite !vassocr ;
           exact (!p)).
      Defined.

      Definition em_comnd_ump_cell_inv_left
        : em_comnd_ump_cell α p • em_comnd_ump_cell_inv = id₂ _.
      Proof.
        use em_comnd_ump_eq.
        - exact (id2 _).
        - rewrite id2_left.
          rewrite id2_rwhisker, id2_right.
          apply idpath.
        - rewrite <- !rwhisker_vcomp.
          unfold em_comnd_ump_cell_inv.
          rewrite !em_comnd_ump_cell_eq.
          apply vcomp_rinv.
        - apply id2_rwhisker.
      Qed.

      Definition em_comnd_ump_cell_inv_right
        : em_comnd_ump_cell_inv • em_comnd_ump_cell α p = id₂ _.
      Proof.
        use em_comnd_ump_eq.
        - exact (id2 _).
        - rewrite id2_left.
          rewrite id2_rwhisker, id2_right.
          apply idpath.
        - rewrite <- !rwhisker_vcomp.
          unfold em_comnd_ump_cell_inv.
          rewrite !em_comnd_ump_cell_eq.
          apply vcomp_linv.
        - apply id2_rwhisker.
      Qed.

      Definition em_comnd_ump_cell_is_invertible
        : is_invertible_2cell (em_comnd_ump_cell α p).
      Proof.
        use make_is_invertible_2cell.
        - exact em_comnd_ump_cell_inv.
        - exact em_comnd_ump_cell_inv_left.
        - exact em_comnd_ump_cell_inv_right.
      Defined.
    End Invertible.

    Definition em_comnd_ump_inv2cell
               {x : B}
               {g₁ g₂ : x --> e}
               (α : invertible_2cell
                      (g₁ · mor_of_em_comnd_cone e)
                      (g₂ · mor_of_em_comnd_cone e))
               (p : α
                    • (g₂ ◃ cell_of_em_comnd_cone e)
                    • lassociator _ _ _
                   =
                   (g₁ ◃ cell_of_em_comnd_cone e)
                   • lassociator _ _ _
                   • (α ▹ h))
      : invertible_2cell g₁ g₂.
    Proof.
      use make_invertible_2cell.
      - exact (em_comnd_ump_cell α p).
      - exact (em_comnd_ump_cell_is_invertible α p).
    Defined.
  End Projections.

  (**
   2. Being an Eilenberg-Moore object for a comonad is a proposition
   *)
  Definition isaprop_has_em_comnd_ump
             (HB : is_univalent_2_1 B)
             (e : em_comnd_cone)
    : isaprop (has_em_comnd_ump e).
  Proof.
    use invproofirrelevance.
    intros φ₁ φ₂.
    use subtypePath.
    {
      intro.
      repeat (use impred ; intro).
      apply isapropiscontr.
    }
    use funextsec ; intro.
    use (path_em_comnd_cone_mor HB).
    - use (em_comnd_ump_inv2cell φ₁).
      + exact (comp_of_invertible_2cell
                 (inv_of_invertible_2cell (inv2cell_of_em_comnd_cone_mor (pr1 φ₁ x)))
                 (inv2cell_of_em_comnd_cone_mor (pr1 φ₂ x))).
      + abstract
          (cbn ;
           rewrite <- !rwhisker_vcomp ;
           rewrite !vassocl ;
           rewrite !(maponpaths (λ z, _ • z) (vassocr _ _ _)) ;
           rewrite <- (em_comnd_cone_mor_endo (pr1 φ₂ x)) ;
           rewrite !vassocr ;
           apply maponpaths_2 ;
           use vcomp_move_R_pM ; [ is_iso | ] ; cbn ;
           rewrite !vassocr ;
           use vcomp_move_L_Mp ; [ is_iso | ] ; cbn ;
           exact (em_comnd_cone_mor_endo (pr1 φ₁ x))).
    - cbn.
      rewrite em_comnd_ump_cell_eq.
      rewrite !vassocr.
      rewrite vcomp_rinv.
      apply id2_left.
  Qed.
End EilenbergMooreComonad.

(**
 3. Bicategories with Eilenberg-Moore objects for comonads
 *)
Definition has_em_comnd
           (B : bicat)
  : UU
  := ∏ (m : mnd (op2_bicat B)),
     ∑ (e : em_comnd_cone m),
     has_em_comnd_ump m e.


(*
   Implementing reals a la Stolzenberg

   Danko Ilik, March 2007

   XField.v -- (unfinished) axiomatisation of the theories of real and
               rational intervals.
*)

Definition associative (A:Type)(op:A->A->A) :=
  forall x y z:A, op (op x y) z = op x (op y z).

Definition commutative (A:Type)(op:A->A->A) :=
  forall x y:A, op x y = op y x.

Definition trichotomous (A:Type)(R:A->A->Prop) :=
  forall x y:A, R x y \/ x=y \/ R y x.

Definition relation (A:Type) := A -> A -> Prop.
Definition reflexive (A:Type)(R:relation A) := forall x:A, R x x.
Definition transitive (A:Type)(R:relation A) :=
  forall x y z:A, R x y -> R y z -> R x z.
Definition symmetric (A:Type)(R:relation A) := forall x y:A, R x y -> R y x.

Record interval (X:Set)(le:X->X->Prop) : Set :=
  interval_make {
    interval_left : X;
    interval_right : X;
    interval_nonempty : le interval_left interval_right
  }.

Record I (grnd:Set)(le:grnd->grnd->Prop) : Type := Imake {
  Icar    := interval grnd le;
  Iplus   : Icar -> Icar -> Icar;
  Imult   : Icar -> Icar -> Icar;
  Izero   : Icar;
  Ione    : Icar;
  Iopp    : Icar -> Icar;
  Iinv    : Icar -> Icar;
  Ic      : Icar -> Icar -> Prop; (* consistency *)
  (* monoids *)
  Iplus_assoc    : associative Icar Iplus;
  Imult_assoc    : associative Icar Imult;
  (* abelian groups *)
  Iplus_comm     : commutative Icar Iplus;
  Imult_comm     : commutative Icar Imult;
  Iplus_0_l      : forall x:Icar, Ic (Iplus Izero x) x;
  Iplus_0_r      : forall x:Icar, Ic (Iplus x Izero) x;
  Imult_0_l      : forall x:Icar, Ic (Imult Ione x) x;
  Imult_0_r      : forall x:Icar, Ic (Imult x Ione) x;
  Iplus_opp_r    : forall x:Icar, Ic (Iplus x (Iopp x)) (Izero);
  Imult_inv_r    : forall x:Icar, ~(Ic x Izero) -> Ic (Imult x (Iinv x)) Ione;
  (* distributive laws *)
  Imult_plus_distr_l : forall x x' y y' z z' z'',
    Ic x x' -> Ic y y' -> Ic z z' -> Ic z z'' ->
    Ic (Imult (Iplus x y) z) (Iplus (Imult x' z') (Imult y' z''));
  (* order and lattice structure *)
  Ilt          : Icar -> Icar -> Prop;
  Ilc          := fun (x y:Icar) => Ilt x y \/ Ic x y;
  Isup         : Icar -> Icar -> Icar;
  Iinf         : Icar -> Icar -> Icar;
  Ilt_trans    : transitive _ lt;
  Ilt_trich    : forall x y:Icar, Ilt x y \/ Ic x y \/ Ilt y x;
  Isup_lub     : forall x y z:Icar, Ilc x z -> Ilc y z -> Ilc (Isup x y) z;
  Iinf_glb     : forall x y z:Icar, Ilc x y -> Ilc x z -> Ilc x (Iinf y z);
  (* order preserves operations? *)
  (* properties of Ic *)
  Ic_refl   : reflexive _ Ic;
  Ic_sym    : symmetric _ Ic
}.

Definition interval_set (X:Set)(le:X->X->Prop) :=
  (interval X le) -> Prop. (* can be Set as well *)
Check interval_set.
Check Ic.
Definition consistent (X:Set)(le:X->X->Prop)(TI:I X le)(p:interval_set X le) :=
  forall I J:interval X le, p I -> p J -> (Ic X le TI) I J.
Check consistent.
(* define 'fine' *)

Record N (grnd:Set)(le:grnd->grnd->Prop)(grndI:I grnd le) : Type := Nmake {
  Ncar    := interval_set grnd le;
  Nplus   : Ncar -> Ncar -> Ncar;
  Nmult   : Ncar -> Ncar -> Ncar;
  Nzero   : Ncar;
  None    : Ncar;
  Nopp    : Ncar -> Ncar;
  Ninv    : Ncar -> Ncar;
  Nc      : Ncar -> Ncar -> Prop; (* Ncistency *)
  (* monoids *)
  Nplus_assoc    : associative Ncar Nplus;
  Nmult_assoc    : associative Ncar Nmult;
  (* abelian groups *)
  Nplus_comm     : commutative Ncar Nplus;
  Nmult_comm     : commutative Ncar Nmult;
  Nplus_0_l      : forall x:Ncar, Nc (Nplus Nzero x) x;
  Nplus_0_r      : forall x:Ncar, Nc (Nplus x Nzero) x;
  Nmult_0_l      : forall x:Ncar, Nc (Nmult None x) x;
  Nmult_0_r      : forall x:Ncar, Nc (Nmult x None) x;
  Nplus_opp_r    : forall x:Ncar, Nc (Nplus x (Nopp x)) (Nzero);
  Nmult_inv_r    : forall x:Ncar, ~(Nc x Nzero) -> Nc (Nmult x (Ninv x)) None;
  (* distributive laws *)
  Nmult_plus_distr_l : forall x x' y y' z z' z'',
    Nc x x' -> Nc y y' -> Nc z z' -> Nc z z'' ->
    Nc (Nmult (Nplus x y) z) (Nplus (Nmult x' z') (Nmult y' z''));
  (* order and lattice structure *)
  Nlt          : Ncar -> Ncar -> Prop;
  Nlc          := fun (x y:Ncar) => Nlt x y \/ Nc x y;
  Nsup         : Ncar -> Ncar -> Ncar;
  Ninf         : Ncar -> Ncar -> Ncar;
  Nlt_trans    : transitive _ lt;
  Nlt_trich    : forall x y:Ncar, Nlt x y \/ Nc x y \/ Nlt y x;
  Nsup_lub     : forall x y z:Ncar, Nlc x z -> Nlc y z -> Nlc (Nsup x y) z;
  Ninf_glb     : forall x y z:Ncar, Nlc x y -> Nlc x z -> Nlc x (Ninf y z);
  (* order preserves operations? *)
  (* properties of Nc *)
  Nc_refl   : reflexive _ Nc;
  Nc_sym    : symmetric _ Nc
}.